

Tabela de conteúdos

1. [Introduction](#) 1.1
2. [Visão geral](#) 1.2
 1. [Configuração](#) 1.2.1
 2. [Opções](#) 1.2.2
 3. [Empacotamento](#) 1.2.3
 4. [Ferramentas](#) 1.2.4
3. [Sintaxe](#) 1.3
 1. [Formatação](#) 1.3.1
 2. [Importações](#) 1.3.2
 3. [Entradas](#) 1.3.3
 4. [Tipos](#) 1.3.4
 1. [Any](#) 1.3.4.1
 2. [Void](#) 1.3.4.2
 3. [Boolean](#) 1.3.4.3
 4. [Number](#) 1.3.4.4
 5. [HugeInt](#) 1.3.4.5
 6. [Text](#) 1.3.4.6
 7. [Method](#) 1.3.4.7
 8. [List](#) 1.3.4.8
 9. [Scope](#) 1.3.4.9
 10. [Error](#) 1.3.4.10
 11. [Chunk](#) 1.3.4.11
 5. [Controle de fluxo](#) 1.3.5
 6. [Loops](#) 1.3.6
4. [Bibliotecas](#) 1.4
 1. [async](#) 1.4.1
 2. [bridge](#) 1.4.2
 3. [color](#) 1.4.3
 4. [console](#) 1.4.4
 5. [curses](#) 1.4.5
 6. [enigma](#) 1.4.6
 7. [failure](#) 1.4.7
 8. [file](#) 1.4.8
 9. [http](#) 1.4.9
 10. [math](#) 1.4.10
 11. [recode](#) 1.4.11
 12. [sdk](#) 1.4.12
 13. [smtp](#) 1.4.13
 14. [system](#) 1.4.14
 15. [time](#) 1.4.15
 16. [type_](#) 1.4.16
 1. [Void](#) 1.4.16.1
 2. [Boolean](#) 1.4.16.2
 3. [Number](#) 1.4.16.3
 4. [HugeInt](#) 1.4.16.4
 5. [Text](#) 1.4.16.5
 6. [Method](#) 1.4.16.6
 7. [List](#) 1.4.16.7
 8. [Scope](#) 1.4.16.8
 9. [Error](#) 1.4.16.9
 10. [Chunk](#) 1.4.16.10
 17. [extra_](#) 1.4.17
 1. [Date](#) 1.4.17.1
 2. [Duration](#) 1.4.17.2
 3. [Fuzzy](#) 1.4.17.3
 4. [HashMap](#) 1.4.17.4
 5. [Logger](#) 1.4.17.5
 6. [Memo](#) 1.4.17.6
 7. [MouseEvent](#) 1.4.17.7
 8. [Opaque](#) 1.4.17.8

- 9. [Option](#) 1.4.17.9
- 10. [Param](#) 1.4.17.10
- 11. [Sound](#) 1.4.17.11
- 12. [Storable](#) 1.4.17.12
- 18. [Comandos embutidos](#) 1.4.18

Introduction



Olá Mundo

```
_ <- fat.std  
console.log('Olá Mundo')
```

Início Rápido

Vá diretamente para a documentação:

- [Visão geral](#)
- [Sintaxe da linguagem](#)
- [Bibliotecas padrão](#)

Interpretador Fry

Para execução local, utilize o interpretador `fry`. É livre e de código aberto! Você pode encontrar o código-fonte, exemplos e mais em nosso [repositório no GitLab](#).

Para detalhes sobre sua instalação e uso, consulte a seção de [configuração](#).

Playground na Web

Para testes rápidos e convenientes, execute seu código diretamente no [FatScript Playground](#). O playground oferece um REPL baseado em web com uma interface intuitiva que permite carregar scripts a partir de um arquivo.

Tutoriais

Mergulhe em nossos tutoriais imersivos, insights por trás dos bastidores e tópicos relacionados no [canal do YouTube FatScript](#).

Download de PDF

- [FatScript v4.0.0 \(atual\)](#)
- [FatScript v3.4.0 \(legado\)](#)
- [FatScript v2.6.0 \(legado\)](#)
- [FatScript v1.3.5 \(legado\)](#)

Doações

Você achou o FatScript útil e gostaria de agradecer?

[Compre-me um café](#)

Licença

[GPLv3](#) © 2022-2025 Antonio Prates

[fatscript.org](#)

Published on Sun Jan 19 2025 21:24:40 GMT+0000 (Greenwich Mean Time)

Visão geral

FatScript é uma linguagem de programação leve e interpretada projetada para criar aplicativos baseados em console. Ela enfatiza a simplicidade, facilidade de uso e conceitos de programação funcional.

Mas, peraí...

FatScript, uma linguagem de programação **leve**?

Sim, há algo estranho nessa afirmação... mas, por favor, deixe-me explicar.

A expressão "açúcar sintático" refere-se a recursos que facilitam a escrita do código, escondendo a complexidade subjacente. E, assim como consumir muito açúcar... pode haver consequências: gordura (fat, em inglês). O que, nesse sentido, é algo positivo - muito peso em poucas linhas de código.

Dito isso, o FatScript ainda é uma linguagem relativamente nova e, embora tenha sido projetada para ser simples e intuitiva, pode não ser a melhor escolha para todas as tarefas, especialmente quando se trata de computação de alto desempenho ou cargas de trabalho extremamente complexas. No entanto, apesar do seu nome, o interpretador do FatScript é pequeno (leve), com um custo de inicialização quase zero e os [benchmarks](#) mostram que ele tem um desempenho comparável ao de linguagens como Python ou JavaScript.

Então, embora chamar o FatScript de "leve" possa ser discutível, o runtime da linguagem não é inerentemente pesado e mantém um perfil eficiente na prática para a maioria dos casos de uso.

Conceitos chave

- Gerenciamento automático de memória por coleta de lixo (GC)
- Combinações simbólicas de caracteres para uma sintaxe minimalista
- REPL (Read-Eval-Print Loop) para testes rápidos de expressões
- Suporte para sistema de tipos, herança e subtipagem por meio de aliases
- Suporte para programação imutável e métodos passáveis (como valores)
- Manter-se simples e intuitivo, sempre que possível

Livre e de código aberto

`fatscript/fry` é um projeto de código aberto que incentiva a colaboração e o compartilhamento de conhecimento. Nós convidamos os desenvolvedores a [contribuir](#) para o projeto e nos ajudar a melhorá-lo com o tempo.

Conteúdo desta seção

- [Configuração](#): como instalar o interpretador de FatScript
- [Opções](#): como personalizar a execução
- [Empacotamento](#): como empacotar um aplicativo FatScript
- [Ferramentas](#): visão geral de algumas ferramentas e recursos extras

Configuração

Para começar a "fritar" seu código "gordo", você precisará de um interpretador para a linguagem de programação FatScript.

fry, O Interpretador FatScript

[fry](#) é um interpretador e ambiente de execução gratuito para FatScript. Você pode instalá-lo em sua máquina seguindo as instruções a seguir.

Instruções

fry é projetado para GNU/Linux, mas também pode funcionar em [outros sistemas operacionais](#).

Para distribuições baseadas em Arch, instale através do pacote AUR [fatscript-fry](#).

Para outras distribuições, experimente o script de instalação automática:

```
curl -sSL https://gitlab.com/fatscript/fry/raw/main/get_fry.sh -o get_fry.sh;
bash get_fry.sh || sudo bash get_fry.sh
```

Ou, para instalar fry manualmente:

- Clone o repositório:

```
git clone --recursive https://gitlab.com/fatscript/fry.git
```

- Depois, execute o script de instalação:

```
cd fry
./install.sh
```

a instalação manual pode copiar o binário fry para a pasta \$HOME/.local/bin, alternativamente, use sudo para instalá-lo em /usr/local/bin/

- Verifique se o fry foi instalado, executando:

```
fry --version
```

Dependências

Se a instalação falhar, podem estar faltando algumas dependências. fry requer git, gcc e libcurl para compilar. Por exemplo, para instalar essas dependências no Debian/Ubuntu, execute:

```
apt update
apt install git gcc libcurl4-openssl-dev
```

Back-end para entrada de texto

linenoise é uma dependência leve e uma alternativa ao readline, mantida como um submódulo. Se não foi incluída durante a operação inicial de git clone, você pode corrigir isso com os seguintes comandos:

```
git submodule init
git submodule update
```

Se você preferir "linkar" com o readline, apenas certifique-se de que ele esteja instalado, executando:

```
apt install libreadline-dev
```

Suporte de Sistema Operacional

fry é primordialmente projetado para GNU/Linux, mas também é acessível em outros sistemas operacionais:

Android

Se você estiver no Android, pode instalar o fry via [Termux](#). Basta instalar as dependências necessárias da seguinte maneira:

Configuração

```
pkg install git clang
```

Em seguida, você pode seguir as instruções padrão de instalação do fry.

ChromeOS

Se você estiver usando o ChromeOS, pode habilitar o suporte ao Linux seguindo as instruções [aqui](#).

MacOS

Se você estiver usando o MacOS, precisará ter as [Command Line Tools](#) instaladas.

iOS

Se você estiver usando o iOS, poderá usar o fry via [iSH](#). Primeiro, instale as dependências necessárias:

```
apk add bash gcc libc-dev curl-dev
```

Em seguida, de acordo com esta [discussão](#), configure o git para funcionar corretamente, assim:

```
wget https://dl-cdn.alpinelinux.org/alpine/v3.11/main/x86/git-2.24.4-r0.apk
apk add ./git-2.24.4-r0.apk
git config --global pack.threads "1"
```

Windows

Se você estiver usando o Windows, poderá usar o fry via [Windows Subsystem for Linux \(WSL\)](#).

Imagem Docker

fry também está disponível como uma [imagem docker](#):

```
docker run --rm -it fatscript/fry
```

Para executar um arquivo FatScript com o docker, use o seguinte comando:

```
docker run --rm -it -v ~/project:/app fatscript/fry prog.fat
```

substitua ~/project pelo caminho para o seu arquivo FatScript

Solução de problemas

Se você encontrar qualquer problema ou bug ao usar o fry, por favor [abra uma "issue"](#).

Opções

Com esta descrição dos modos e parâmetros disponíveis, você descobrirá que o `fry` tem várias especiarias guardadas na manga para você temperar a execução do seu código.

Argumentos de linha de comando

A interface CLI oferece alguns modos de operação:

- `fry [OPÇÕES]` read-eval-print-loop (REPL)
- `fry [OPÇÕES] ARQUIVO [ARGS]` executar um arquivo FatScript
- `fry [OPÇÕES] -b/-o ENTRADA SAÍDA` criar um bundle
- `fry [OPÇÕES] -f ARQUIVO...` formatar arquivos de código-fonte do FatScript

Aqui estão os parâmetros de opção disponíveis:

- `-a`, `--ast` exibir apenas a árvore sintática abstrata
- `-b`, `--bundle` salvar bundle em arquivo de saída
- `-c`, `--clock` registro de tempo e estatísticas (alternar)
- `-d`, `--debug` habilitar logs de debug (implica em `-c`)
- `-e`, `--error` continuar em caso de erro (alternar)
- `-f`, `--format` indentar arquivos de código-fonte do FatScript
- `-h`, `--help` exibir ajuda e sair
- `-i`, `--interactive` habilitar REPL com execução do arquivo
- `-j`, `--jail` restringir sistema de arquivos, rede e chamadas de sistema
- `-k`, `--stack #` definir a profundidade da pilha (contagem de frames)
- `-m`, `--meta` exibir informações sobre a build
- `-n`, `--nodes #` definir limite de memória (contagem de nós)
- `-o`, `--obfuscate` ofuscar o bundle (implica em `-b`)
- `-p`, `--probe` realizar análise estática (teste seco)
- `-s`, `--save` armazenar sessão do REPL em `repl.fat`
- `-v`, `--version` exibir número da versão e sair
- `-w`, `--warranty` exibir isenção de responsabilidade e sair

a opção `-e` é ativada automaticamente nos modos REPL e probe

combinar `-p` com `-f` envia o resultado formatado para stdout

combinar `-p` com `-b` imprime a análise de código ao empacotar

Gerenciamento de memória

`fry` gerencia a memória automaticamente sem pré-reserva. Você pode limitar o uso da memória especificando o número de nós com as opções da CLI:

- `-n <count>` para uma contagem exata de nós
- `-n <count>k` para kilonós, contagem * 1000
- `-n <count>m` para meganós, contagem * 1000000

Por exemplo, `fry -n 5k meuPrograma.fat` restringe o aplicativo a 5000 nós.

O coletor de lixo (GC) é executado automaticamente quando restam 256 nós antes que o limite final de memória seja atingido (premonição do GC, ver `parameters.h`). Você também pode invocar o GC a qualquer momento chamando o método `runGC` da [biblioteca SDK](#) desde a thread principal.

Estimativa de bytes (x64)

Cada nó em uma plataforma de 64 bits usa aproximadamente ~200 bytes. O tamanho real do nó depende dos dados que ele contém. Por exemplo, o limite padrão é 10 milhões de nós, seu programa pode chegar a usar cerca de 2 GB de RAM ao atingir o limite padrão.

Use a opção `-c` ou `--clock` para imprimir as estatísticas de execução e ter uma melhor compreensão de como seu programa está se comportando na prática.

Verificação de tempo de execução

Existem dois [comandos embutidos](#) para verificar o uso de memória em tempo de execução:

- `$nodesUsage` nós alocados no momento
- `$bytesUsage` máximo de bytes alocados (`ru_maxrss`)

Tamanho da pilha

A profundidade máxima da pilha é definida em `parameters.h`, no entanto, você pode personalizar o tamanho da pilha até certo ponto usando opções da linha de comando (CLI):

- `-k <count>` para um número exato de frames
- `-k <count>k` para kiloframes, `count * 1000`

Arquivo "run commands"

Na inicialização, `fry` procura um arquivo `.fryrc` no mesmo caminho do arquivo do programa e, se não encontrado, também no diretório atual. Se encontrado, é executado como uma fase de "pré-cozimento" para configurar o ambiente para a execução do programa.

Gerenciamento de memória com `.fryrc`

Você pode usar o arquivo `.fryrc` para definir o limite de memória para seu projeto sem precisar especificá-lo como argumento da CLI. Para fazer isso, você pode usar o método `setMem` fornecido pela [biblioteca SDK](#), assim:

```
_ <- fat.sdk
setMem(64000) # define 64k nós como limite de memória
```

Detalhes de inicialização

As opções da linha de comando são aplicadas primeiro, exceto pelo limite de memória. Durante a fase de pré-cozimento, o `fry` usa o limite padrão de 10 milhões de nós, independentemente da opção da linha de comando. Se você definir um limite de memória no arquivo `.fryrc`, esse limite terá efeito a partir desse ponto e substituirá a opção da linha de comando para toda a execução. Se o arquivo `.fryrc` não definir um limite de memória, a opção da linha de comando terá efeito após a fase de pré-cozimento.

O escopo de pré-cozimento é invisível por padrão. Após a execução do arquivo `.fryrc`, um escopo zerado é fornecido para o seu programa, o que permite testar seu código com um limite muito baixo de nós ao usar um arquivo `.fryrc` sem afetar a contagem de nós. Isso também impede que o namespace `.fryrc` entre em conflito com o escopo global do seu programa. No entanto, se você quiser manter as entradas declaradas no `.fryrc` no escopo global para fins de configuração, pode chamar o comando embutido `$keepDotFry` em algum lugar do arquivo `.fryrc`.

Outro uso possível, além de configurar o limite de memória, é pré-carregar as importações comuns, por exemplo, os tipos padrão:

```
$keepDotFry
_ <- fat.type._
```

Modo Sandbox

Use a opção `-j` ou `--jail` para inibir os seguintes comandos embutidos:

- `write`, `remove` e `mkdir` - Esses comandos modificam o sistema de arquivos.
- `request` - Este comando é usado para fazer requisições HTTP externas.
- `send` - Este comando é usado para enviar e-mails via SMTP.
- `loadDLL` - Este comando carrega uma biblioteca externa via `dlopen`.
- `shell`, `capture`, `fork` e `kill` - Estes comandos estão envolvidos em iniciar ou parar processos arbitrários.

Veja também

- [Comandos embutidos](#)
- [Biblioteca SDK](#)

Empacotamento

O fry oferece uma ferramenta integrada de empacotamento para código FatScript.

Utilização

Para agrupar seu projeto em um único arquivo a partir do ponto de entrada, execute:

```
fry -b sweet mySweetProject.fat
```

A seguir, você pode executar seu programa:

```
./sweet
```

Este processo faz o seguinte:

- Consolida todas as importações, exceto as bibliotecas padrão e [caminhos literais](#)
- Remove espaços e comentários para melhorar os tempos de carregamento
- Substitui quaisquer instruções `$break` (ponto de interrupção do depurador) por `()`
- Adiciona um [shebang](#) ao código empacotado
- Recebe o atributo de execução para o modo de arquivo

Considerações

As importações são deduplicadas e incluídas com base na ordem de sua primeira aparição. Como resultado, a sequência em que você importa seus arquivos desempenha um papel crítico no resultado final agrupado. Embora essas considerações geralmente sejam inconsequentes para projetos pequenos, o empacotamento de projetos maiores pode exigir uma organização adicional. Sempre valide o seu código empacotado.

Ofuscação

Para uma ofuscação opcional, use `-o`:

```
fry -o sweet mySweetProject.fat # cria o pacote ofuscado
./sweet                        # executa seu programa da mesma maneira
```

ao distribuir por meio de hosts públicos, considere [definir uma chave personalizada](#) com um `.fryrc` local; Apenas o cliente deve ter acesso a esta chave para proteger o fonte

A ofuscação usa o algoritmo [enigma](#) para encriptação, garantindo uma decodificação rápida. Para um tempo de carregamento ótimo, prefira `-b` se a ofuscação não for essencial.

Ferramentas

Aqui estão algumas dicas que podem melhorar sua experiência de programação com FatScript.

Análise estática

Use o modo de verificação para checar a sintaxe e receber dicas sobre o seu código:

```
fry -p mySweetProgram.fat
```

Depurador

Um ponto de interrupção, indicado pelo comando `$break`, atua como uma ferramenta de depuração ao interromper temporariamente a execução do programa em um local designado e carregar o console de depuração integrado. Ele oferece um ambiente interativo para examinar o estado atual do programa, inspecionando valores no escopo, avaliando expressões e rastreando o fluxo do programa.

Para ativar os pontos de interrupção, é necessário executar o programa com modo interativo habilitado:

```
fry -i mySweetProgram.fat
```

No FatScript, `$break` retorna `null`, o que pode alterar um valor de retorno se colocado no final de um bloco, devido ao recurso de [auto-retorno](#). Tenha cuidado com o posicionamento de `$break` para evitar efeitos não intencionais na funcionalidade do programa.

Gerenciador de pacotes

`chef` é o gerenciador de pacotes oficial para FatScript, projetado para fácil gerenciamento de dependências.

Para instalar, clone o repositório e construa `chef`:

```
git clone https://gitlab.com/fatscript/chef.git
cd chef
fry -b $HOME/.local/bin/chef chef.fat
```

Para instruções de uso e mais detalhes, visite o [repositório chef](#).

Formatação do código-fonte

Suporte nativo

Você pode aplicar a indentação automática ao seu código fonte usando o seguinte comando:

```
fry -f mySweetProgram.fat
```

Extensão do Visual Studio Code

Para adicionar suporte de formatação de código ao VS Code, você pode instalar a extensão [fatscript-formatter](#). Abra o Quick Open do VS Code (Ctrl+P), cole o seguinte comando e pressione enter:

```
ext install aprates.fatscript-formatter
```

o `fry` precisa estar instalado em seu sistema para que essa extensão funcione

Realce de sintaxe

Extensão do Visual Studio Code

Para adicionar destaque de sintaxe do FatScript ao VS Code, você pode instalar a extensão [fatscript-syntax](#). Abra o Quick Open do VS Code (Ctrl+P), cole o seguinte comando e pressione enter:

```
ext install aprates.fatscript-syntax
```

Você também pode encontrar e instalar essas extensões no Marketplace de Extensões do VS Code.

Plugin para Vim e Neovim

Para instalar o realce de sintaxe do FatScript para Vim e Neovim, confira o plugin [vim-syntax](#).

Para usuários do Neovim, adicione a linha respectiva à sua configuração:

Usando packer.nvim:

```
use { 'https://gitlab.com/fatscript/vim-syntax', as = 'fatscript' }
```

Usando lazy.nvim:

```
{ 'https://gitlab.com/fatscript/vim-syntax', name = 'fatscript' }
```

Arquivo de sintaxe do Nano

Para instalar o realce de sintaxe do FatScript no nano, siga estes passos:

1. Baixe o arquivo `fat.nanorc` [daqui](#).
2. Copie o arquivo `fat.nanorc` para o diretório de sistema do nano:

```
sudo cp fat.nanorc /usr/share/nano/
```

Se o realce de sintaxe não for habilitado automaticamente, talvez você precise habilitá-lo explicitamente em seu arquivo `.nanorc`. Consulte as instruções na [Wiki do Arch Linux](#) para mais informações.

Após a instalação do destaque de sintaxe, você também pode usar o formatador de código no nano com a seguinte sequência de atalhos:

- Ctrl+T Executar; e em seguida...
- Ctrl+O Formatador

Sintaxe

Aspectos essenciais

Importações (<-)

```
console <- fat.console
```

Valores (v)

Os nomes dos valores começam com letra minúscula:

```
name = 'Mary'
age = 25
```

valores são constantes, salvo que declarados inicialmente com til

Variáveis (~)

```
~ email = 'my@email.com'
~ isOnline = true
```

Listas ([])

```
list = [ 1, 2, 3 ]

list()          # Retorna 1, somente leitura
list[0]         # Retorna 1, leitura/escrita, caso list possa ser alterada
```

Escopos ({})

```
scope = { key1 = 'value1', key2 = 'value2' }

scope.key1      # Retorna 'value1' (acesso por ponto)
scope('key1')  # Retorna 'value1', somente leitura (acesso por chamada)
scope['key1']   # Outputs 'value1', leitura/escrita, caso valor possa ser alterado
```

Tipos (T)

Os nomes dos tipos começam com letra maiúscula:

```
Person = (name: Text, age: Number)
person = Person('Mary', 25)
```

Métodos (->)

```
greeting = (name: Text): Text -> 'Olá, {name}'
console.log(greeting('Mundo'))
```

métodos também são considerados valores

Coalescência nula (??)

```
maybeValue ?? fallback    # usa fallback se maybeValue for null/erro
```

If-Else (?:_)

```
condition ? then : else    # se condition for verdadeira, faz "then", senão "else"
```

Match de casos (=>)

```
condition1 => result1
condition2 => result2
conditionN => resultN
_          => default    # caso geral
```

Switch (>>)

```
value >> {
  match1 => result1
  match2 => result2
  matchN => resultN
  _      => default      # caso geral
}
```

Tap (<<)

```
expression << tapMethod
```

usa tapMethod apenas pelos seus efeitos sobre valor retornado por expression

Loops (@)

```
condition @ loopBody          # loop enquanto condition for verdadeira
1..10 @ n -> rangeMapper(n)   # itera sobre o intervalo de 1 a 10
list @ item -> listMapper(item) # itera sobre os itens da lista
scope @ key -> scopeMapper(key) # itera sobre as chaves do escopo
```

Procedimentos (<>)

```
~ users = [
  { name = 'Foo', age = 30 }
  { name = 'Bar', age = 28 }
]
userNames = List <> users @ -> _.name
userNames # Retorna ['Foo', 'Bar']
```

Exploração detalhada

Nas próximas páginas, você encontrará informações sobre os aspectos centrais da escrita de código FatScript, usando tanto os recursos básicos da linguagem quanto os recursos avançados do sistema de tipos e bibliotecas padrão.

- [Formatação](#): como formatar código FatScript corretamente
- [Importações](#): como importar bibliotecas no seu código
- [Entradas](#): entendendo o conceito de entradas e escopos
- [Tipos](#): um guia para o sistema de tipos do FatScript
 - [Any](#) - qualquer coisa
 - [Void](#) - nada
 - [Boolean](#) - primitivo
 - [Number](#) - primitivo
 - [HugeInt](#) - primitivo
 - [Text](#) - primitivo
 - [Method](#) - função ou lambda
 - [List](#) - como uma matriz ou pilha
 - [Scope](#) - como um objeto ou dicionário
 - [Error](#) - sim, para erros
 - [Chunk](#) - dados binários
- [Controle de fluxo](#): controlando a execução do programa com condicionais
- [Loops](#): como usar intervalos, map-over e loops while

Formatação

No FatScript, espaços em branco e indentação são irrelevantes, porém são muito bem-vindos para tornar o código mais legível e fácil de entender.

Espaços em branco

- Um caractere de nova linha (`\n`) indica o final de uma expressão, exceto quando:
 - o último token na linha é um operador
 - o primeiro token da próxima linha é um operador não-unário
 - usando parênteses para agrupar expressões
- Expressões podem estar na mesma linha se separadas por vírgula (,) ou ponto-e-vírgula (;)

Comentários

Comentários começam com `#` e são terminados por uma nova linha:

```
a = 5 # este é um comentário
```

Nota

FatScript não suporta comentários multi-linhas no momento. Além disso, literais de texto podem acabar como um valor de retorno válido se deixados como a última linha restante, devido à funcionalidade de [auto-retorno](#). Portanto, é recomendável se ater ao formato de comentário de linha única.

Veja também

- [Autoformatador de código fonte](#)

Importações

Vamos desvendar a arte de importar arquivos e bibliotecas no FatScript! Por quê? Bem, porque nesta linguagem você pode importar sempre que seu coração desejar, simplesmente usando uma seta para a esquerda <-.

Sintaxe de ponto

Para usar importações com sintaxe de ponto, os nomes dos arquivos e pastas do projeto não devem começar com dígito nem conter símbolos.

você pode especificar qualquer caminho que desejar usando [caminhos literais](#)

Importação nomeada

Para importar arquivos, use a extensão `.fat` para nomes de arquivo (ou nenhuma extensão). No entanto, omita a extensão na declaração de importação. Aqui está um exemplo:

```
ref <- nomeDoArquivo
```

se ambos os arquivos `x` e `x.fat` existirem, o último terá precedência

Para importar arquivos de pastas:

```
ref1 <- pasta.nomeDoArquivo
ref2 <- pasta.subPasta.nomeDoArquivo
```

Para importar todos os arquivos de uma pasta, use a sintaxe de "ponto-sublinhado":

```
lib <- pasta._
```

Observe: apenas os arquivos imediatamente dentro da pasta são incluídos usando a sintaxe acima. Para incluir arquivos de subpastas, mencione-os explicitamente. Além disso, um arquivo `"_fat"` (ou arquivo `"_"`) dentro de uma pasta pode substituir o comportamento de importação de "ponto-sublinhado".

barras / também podem ser usadas como alternativa, como em `ref <- folder/filename`

Acesso de elementos

Uma vez importado, acesse os elementos usando a sintaxe de ponto:

```
ref1.elemento1
```

Extração de elementos

Para extrair elementos específicos de uma importação nomeada ou para evitar adicionar o nome do módulo todas as vezes (por exemplo, `lib.foo`), use [atribuição por desestruturação](#):

```
{ foo, bar } = lib
```

Visibilidade

Importações nomeadas são resolvidas no escopo global, independentemente de onde forem declaradas. Isso significa que mesmo se você declarar uma importação nomeada dentro de uma função ou escopo local, ela será globalmente acessível.

Importação local

Para importar no escopo atual, use:

```
_ <- nomeDoArquivo
```

Importações locais, ao contrário das nomeadas, despejam o conteúdo do arquivo diretamente no escopo atual. Assim, um método importado pode ser invocado como `baz(arg)` em vez de `ref.baz(arg)`.

Embora as importações locais sejam bem apropriadas para importar [tipos](#) no escopo global, elas devem ser usadas com cautela ao importar conteúdos de biblioteca. O uso excessivo de importações locais pode levar a poluição do namespace, tornando mais

desafiador seguir o código, porque fica menos aparente de onde vêm os métodos.

Caminhos literais

Com caminhos literais, você pode usar qualquer nome de arquivo ou extensão. No entanto, observe que essas importações não são avaliadas durante o [empacotamento](#), mas em tempo de execução. Aqui está um exemplo:

```
ref <- '_pasta/fonte-2.outro'
```

Você também pode usar [textos inteligentes](#) como caminhos literais:

```
base = 'pasta'  
arquivo = 'fonte.xyz'  
ref <- '{base}/{arquivo}'
```

Lembre-se de que caminhos literais podem tornar seu código mais complexo e essas importações só podem ser resolvidas dinamicamente, então use-os com moderação.

Deduplicação de importação

O FatScript utiliza uma estratégia de "importar uma vez" com um mecanismo de sinalizador no escopo, ignorando automaticamente os arquivos que já foram importados.

Armadilhas do uso de importação

1. **Importações locais dentro do método:** importar diretamente dentro de um corpo de método reavalia a importação em cada invocação, causando retenção de memória:

```
myMethod = -> {  
  _ <- lib # vazamento de memória potencial  
  ...  
}
```

Esse comportamento não é classificado como um bug em si, mas sim uma consequência das escolhas de design no sistema de coleta de lixo (GC) do FatScript. As otimizações do GC excluem nós derivados diretamente do código-fonte, permitindo que eles evitem procedimentos padrão de marcação e varredura. Como resultado, as importações locais dentro dos métodos perdem a oportunidade de deduplicação, fazendo com que seus nós permaneçam residentes até o fim do programa.

2. **Importações locais seletivas:** Usar atribuição de desestruturação em importações locais descarta outros membros, mas toda a importação é processada e vinculada ao contexto do membro extraído:

```
{ foo1 } = { _ <- lib } # lib é carregada e vinculada ao contexto de foo1  
...  
{ foo2 } = { _ <- lib } # lib é carregada novamente e vinculada a foo2
```

Este padrão cria um fechamento. Usá-lo para a mesma biblioteca resulta em cargas repetidas, aumentando o uso de memória. Para melhor eficiência, considere importar a biblioteca uma vez no nível superior e referenciá-la diretamente ou usar importações seletivas com moderação.

Melhores práticas

Para evitar problemas de memória, siga estas estratégias:

- **Mover importações para escopo externo:** Importar bibliotecas em um nível mais alto para garantir avaliação única.
- **Usar importações nomeadas:** Prefira importações nomeadas para reutilizar código sem redundância.

Entradas

As entradas são pares chave-valor que existem no escopo onde são declaradas.

Nomeação

Os nomes das entradas (chaves) **não** podem começar com uma letra maiúscula, que é a distinção com relação aos [tipos](#). Os identificadores são sensíveis a maiúsculas e minúsculas, portanto "batatasfritas" e "batatasFritas" seriam considerados entradas diferentes.

A convenção recomendada é usar `camelCase` para as entradas.

você pode usar um nome arbitrário como chave usando [nomeação dinâmica](#)

Declaração e atribuição

Em FatScript, você pode declarar entradas simplesmente atribuindo um valor:

```
isOnline: Boolean = true
age: Number      = 25
name: Text       = 'João'
```

Os tipos também podem ser inferidos a partir da atribuição:

```
isOnline = true    # Boolean
age      = 25      # Number
name     = 'João'  # Text
```

Entradas imutáveis

No FatScript, ao declarar uma entrada, ela é por padrão imutável, o que significa que, uma vez atribuído, seu valor não pode ser alterado. Essa imutabilidade garante consistência ao longo da execução do programa:

```
fruta = 'banana'
fruta = 'abacate' # gera um erro porque 'fruta' é imutável
```

Exceção à Regra

A imutabilidade no FatScript aplica-se à vinculação da entrada, não ao conteúdo de escopos. Embora uma entrada seja imutável, se ela contiver um escopo, o conteúdo desse escopo pode ser modificado, seja pela adição de novas entradas, ou pela alteração de entradas mutáveis contidas no escopo:

```
s = { a = 1, b = 2 }
s.c = 3 # mesmo que 's' seja imutável, ele aceita o novo valor de 'c'
s      # agora é { a = 1, b = 2, c = 3 }
```

Essa escolha de design oferece flexibilidade com modificações de escopo. Em contraste, [listas](#) impõem uma imutabilidade mais estrita, impedindo a adição de novas entradas a listas imutáveis.

Escopos são sempre passados por referência. Para modificar o conteúdo de um escopo sem alterar sua referência original, use o método `copy` da [extensão de protótipo de Escopo](#) para criar uma duplicata.

Selagem de Escopos

Começando com a versão 4.x.x, a [extensão de protótipo de Escopo](#) introduz o método `seal`, permitindo que você evite modificações futuras em um escopo selando-o. Uma vez selado, nenhuma nova entrada pode ser adicionada ao escopo, embora entradas existentes ainda possam ser modificadas (se mutáveis):

```
s = { ~ a = 1, b = 2 }
s.seal # sela o escopo
s.c = 3 # gera um erro: não é possível adicionar novos membros a um escopo selado
s.a = 42 # permitido: modifica uma entrada mutável existente
```

Entradas mutáveis

Entradas

Sim, você pode declarar entradas mutáveis, também conhecidas como variáveis. Para declarar uma entrada mutável, use o operador til ~:

```
~ fruta = 'banana'
fruta   = 'abacate' # ok
```

Observe que mesmo uma entrada mutável não pode mudar imediatamente seu tipo, a menos que seja apagada do escopo. Para apagar uma entrada, atribua `null` a ela e, em seguida, redeclare-a com um novo tipo. Mudar tipos é desencorajado pela sintaxe e não é recomendado, mas é possível:

```
~ color = 32 # cria a entrada color como um número mutável
color = 'blue' # gera um TypeError porque color é um número
color = null # a entrada é apagada
color = 'blue' # redefine color com um tipo diferente (Texto)
```

você deve declarar a entrada como mutável novamente usando o til ~ ao redefini-la após a exclusão se quiser que o próximo valor seja mutável

Entradas dinâmicas

Você pode criar entradas com nomes dinâmicos usando colchetes [ref]:

```
ref = 'pipoca' # texto será o nome da entrada
```

```
opcoes = { [ref] = 'é saborosa' }
```

```
opcoes[ref] # sintaxe dinâmica: 'é saborosa', com acesso de leitura e gravação
opcoes(ref) # sintaxe de obtenção: 'é saborosa', mas o valor é somente leitura
opcoes.pipoca # sintaxe de ponto: 'é saborosa', mas deve seguir a nomenclatura
```

todas as declarações dinâmicas são entradas mutáveis

Essa funcionalidade permite definir dinamicamente os nomes dentro de um escopo e criar entradas com nomes que, de outra forma, não seriam aceitos pelo FatScript.

As entradas dinâmicas também podem usar referências numéricas, mas a referência é convertida em texto automaticamente, e.g.:

```
[ 5 ] = 'texto armazenado na entrada 5'
self['5'] # devolve 'texto armazenado na entrada 5'
self[5] # devolve 'texto armazenado na entrada 5'
```

Atribuição por desestruturação

Você pode copiar os valores de um escopo em outro escopo assim:

```
_ <- fat.math
distancia = (posicao: Scope/Number): Number -> {
  { x, y } = posicao # atribuição por desestruturação no escopo do método
  sqrt(x ** 2 + y ** 2) # calcula a distância entre a origem e (x, y)
}
distancia({ x = 3, y = 5 }) # 5.83095189485
```

A mesma sintaxe funciona de forma semelhante para listas:

```
distancia = (posicao: List/Number): Number -> {
  { x, y } = posicao # extrai os dois primeiros itens para 'x' e 'y'
  sqrt(x ** 2 + y ** 2)
}
distancia([ 3, 5 ]) # 5.83095189485
```

Você também pode usar a atribuição por desestruturação para expor um determinado método ou propriedade de uma [importação nomeada](#):

```
console <- fat.console
{ log } = console
log('Olá Mundo')
```

usando essa sintaxe com importações, você pode escolher trazer para o escopo atual apenas os elementos da biblioteca que você está interessado em usar, evitando assim a poluição do namespace com nomes que não teriam uso ou poderiam entrar

em conflito com os de sua própria autoria

Sintaxe semelhante a JSON

O FatScript também suporta sintaxe semelhante a JSON para declarar entradas:

```
"nada": null,                # entrada Void - comportamento distinto, veja abaixo
"estaOnline": true,         # entrada Boolean
"idade": 25,                # entrada Number
"nome": "João",             # entrada Text
"tags": [ "a", "b" ],      # entrada List
"opcoes": { "prop": "outra" } # entrada Scope
```

É importante observar que as declarações semelhantes a JSON sempre criam entradas imutáveis, portanto você não pode adicionar o caractere til ~ para torná-las mutáveis.

Tipos

Os tipos são usados no FatScript para combinar dados e comportamentos, atuando como modelos para a criação de novas instâncias.

Nomeação

Os nomes de tipo são sensíveis a maiúsculas e minúsculas, devendo começar com uma letra maiúscula.

A convenção recomendada para identificadores de tipo é `PascalCase`.

Tipos Nativos

O FatScript fornece vários tipos nativos:

- [Any](#) - qualquer coisa
- [Void](#) - nada
- [Boolean](#) - primitivo
- [Number](#) - primitivo
- [HugeInt](#) - primitivo
- [Text](#) - primitivo
- [Method](#) - função ou lambda
- [List](#) - como uma matriz ou pilha
- [Scope](#) - como um objeto ou dicionário
- [Error](#) - sim, para erros
- [Chunk](#) - dados binários

No entanto, é necessário importar o [pacote type](#) para acessar os membros de protótipo de cada tipo.

Tipos Adicionais

Os tipos nativos do FatScript são enriquecidos com uma coleção de [tipos extras](#) que expandem as funcionalidades básicas de seus tipos nativos. Criados em FatScript puro, esses tipos adicionais atendem a várias necessidades de programação avançada e facilitam padrões de projeto comuns.

Além disso, você encontrará tipos de domínio específico incorporados nas bibliotecas, como `Worker` na biblioteca [async](#), `FileInfo` em [file](#), `HttpRequest` (entre outros) em [http](#), `CommandResult` em [system](#) etc.

Tipos Personalizados

Além de usar os tipos fornecidos pela linguagem ou por uma biblioteca externa, você também pode criar seus próprios tipos ou estender os existentes com novos comportamentos.

Declaração

Para definir um tipo personalizado no FatScript, você pode usar uma simples declaração de atribuição. A definição de tipo pode ser envolvida em parênteses ou chaves. Ambas as sintaxes são válidas e têm o mesmo efeito. Você também pode opcionalmente definir valores padrão para as propriedades do tipo, como mostrado no seguinte exemplo:

```
# Definição de tipo com valores padrão
Carro = (km: Number = 0, cor: Text = 'branco', opcional = null)
```

Unicidade Global

O FatScript possui um único meta-espço global, exigindo que os nomes dos tipos sejam únicos em todo o seu programa e em quaisquer bibliotecas incluídas. Tentar definir um tipo que compartilha um nome com um tipo existente, mesmo que em um escopo diferente, aciona um `AssignError`. No entanto, se a nova definição for idêntica à existente, ela será simplesmente ignorada.

Para examinar os tipos presentes no meta-espço global, o comando `_<-fat.std; sdk.getTypes`; se mostra útil. Esta função enumera todos os tipos definidos e detalha seus locais de definição com marcadores `source:line:column`. Este recurso ajuda na navegação e compreensão da estrutura do seu código e suas dependências.

Tipos

É prudente evitar nomes já utilizados pelos tipos de biblioteca `fat . std` ao definir novos tipos.

Embora o FatScript não imponha um protocolo estrito de nomes para o desenvolvimento de bibliotecas, é recomendado adotar uma estratégia de nomes que evite conflitos. Uma prática comum envolve prefixar os nomes dos tipos com algum identificador único que reflita o nome da sua biblioteca, reduzindo assim a probabilidade de choques de nomes.

Uso

Para criar instâncias de um tipo personalizado, chame o nome do tipo como se fosse um [método](#), opcionalmente passando valores para as propriedades:

```
# Uso do tipo com padrões
carro = Carro()
# saída: { km: Number = 0, cor: Text = 'branco' }

# Uso do tipo, definindo uma das propriedades
carroVermelho = Carro(cor = 'vermelho')
# saída: { km: Number = 0, cor: Text = 'vermelho' }

# Uso do tipo, totalmente qualificado
carroVelho1 = Carro(cor = 'azul', km = 38000)
# substituí ambos os valores

# Uso do tipo, argumentos usando a sequência das propriedades
carroVelho2 = Carro(41000, 'verde')
# substituí valores usando a ordem da definição do tipo
```

Por padrão, os tipos personalizados retornam um escopo de suas propriedades. No entanto, se você definir um procedimento `apply`, o tipo poderá retornar um valor diferente. Por exemplo, aqui está um tipo personalizado `Soma` com um procedimento `apply` que retorna a soma de suas propriedades `a` e `b`:

```
Soma = (a: Number, b: Number, apply = <> a + b)
Soma(1, 2) # saída: 3
```

observe que os procedimentos `apply` têm acesso direto às propriedades da instância

Neste exemplo, o tipo base de saída do `apply` é um número, não um escopo. Isso também significa que as propriedades originais do tipo personalizado são perdidas durante a instanciação e não podem ser acessadas novamente.

Membros do protótipo

Esses são um tipo especial de método, armazenados dentro da definição do tipo:

```
TipoComMembrosDePrototipo = {
  ~ a: Number
  ~ b: Number

  setA = (novoA: Number) -> self.a = novoA
  setB = (novoB: Number) -> self.b = novoB
  soma = (): Number -> self.a + self.b
}
```

Neste exemplo, `setA`, `setB` e `soma` são membros do protótipo. Observe que precisamos usar `self`, que é uma palavra-chave que fornece uma referência ao escopo da própria instância (ou método), para que nós pudéssemos ganhar acesso às propriedades.

Checando tipos

Se você não sabe qual é o tipo de uma entrada, pode simplesmente verificar comparando com um nome de tipo:

```
lugar = 'restaurante'
lugar == Number # false
lugar == Text   # true
```

alternativamente, use o método `typeof` da [biblioteca SDK](#) para extrair o nome do tipo

Qualquer coisa pode ser comparada com a palavra reservada `Type`, que identifica se se refere a um tipo:

```
Number == Type # true
```

Tipos

Type também pode ser usado para especificar que um método recebe um parâmetro de tipo:

```
combine = (t: Type, val: Any): Any -> ...
```

Alias de tipo

No FatScript, você pode criar subtipos atribuindo um nome diferente a um tipo existente. Isso significa que o novo tipo herdará todas as propriedades do tipo base. Aqui está um exemplo:

```
_ <- fat.type.Text
Id = Text # cria um alias
```

Os aliases de tipo são hierárquicos e podem ser usados para classificar valores enquanto ainda herdarem o mesmo comportamento. No entanto, embora o alias seja considerado igual ao tipo base, as instâncias do novo tipo não são consideradas iguais ao tipo base.

Para verificar se um valor é uma instância de um alias de tipo ou do tipo base, você pode usar o operador de comparação de menor-ou-igual <=. Isso permite que você aceite qualquer tipo na cadeia de aliases, até o tipo base. Aqui está um exemplo:

```
Id == Text # verdadeiro, já que Id é um alias de Text
x = Id(123) # id: Id = '123'
x == Text # falso, no entanto x é do tipo Id, não Text
x == Id # verdadeiro, como o esperado x é do tipo Id
x <= Text # verdadeiro, já que x é do tipo Id, que é um alias de Text
```

Essa funcionalidade permite uma validação refinada em tipos específicos, mantendo a flexibilidade de usar diferentes aliases para o mesmo tipo subjacente.

limitação: não é possível criar alias para Any, Type ou Method

Restrições de tipo

No FatScript, você pode declarar restrições de tipo para parâmetros de método. Quando um método é chamado, o argumento é verificado automaticamente em relação à restrição de tipo. Se o argumento não for do tipo esperado ou um de seus subtipos, um `TypeError` é gerado.

Se a restrição de tipo for um tipo base, qualquer subtipo desse tipo também será aceito como argumento. No entanto, se a restrição de tipo for um subtipo, somente argumentos que correspondam ao subtipo serão aceitos. Aqui está um exemplo:

```
generalista = (x: Text) -> x
restritivo = (x: Id) -> x
```

Neste exemplo, o método `generalista` aceita argumentos `Text` e `Id`, porque `Id` é um subtipo de `Text`. O método `restritivo` aceita apenas argumentos `Id` e não `Text`, porque `Id` é um subtipo de `Text`, mas não o contrário.

É importante enfatizar que os tipos personalizados são derivados de `Scope`. Nesse contexto, `Scope` seria o tipo `generalista` para, por exemplo, o tipo personalizado `Carro`.

Mixin (avançado)

Ao definir um tipo, você pode adicionar os recursos de um tipo existente simplesmente mencionando-o na definição de tipo. Isso é chamado de inclusão de tipo ou mixin.

Por exemplo, para criar um novo tipo `CarroAlugado` com as propriedades de `Carro` e uma propriedade adicional `preco`, você pode escrever:

```
CarroAlugado = {
  # Inclusões
  Carro

  # Propriedade adicional
  preco: Number
}

CarroAlugado(50) # { cor: Text = 'branco', km: Number = 0, preco: Number = 50 }
```

Se uma propriedade não estiver definida no novo tipo, ela herdará o valor padrão do tipo incluído. No exemplo acima, as propriedades `cor` e `km` do `Carro` estão presentes no `CarroAlugado`, com seus valores padrão.

Herdando métodos de protótipo

Suponha que continuemos a partir do exemplo anterior do tipo `TipoComMembrosDePrototipo` que tem duas propriedades `a` e `b`, e três métodos de protótipo `setA`, `setB` e `soma`. Para criar um novo tipo `ComMaisMembros` que adiciona uma propriedade `c`, um método `setC` e substitui o método `soma`, você pode escrever:

```
ComMaisMembros = {
  # Inclusões
  TipoComMembrosDePrototipo

  # Propriedades (parâmetros da instância)
  ~ a: Number
  ~ b: Number
  ~ c: Number

  # Membros de protótipo (métodos)
  setC = (novoC: Number) -> self.c = novoC
  soma = (): Number      -> self.a + self.b + self.c
}
```

redeclarando as propriedades permite que o novo tipo também aceite argumentos no momento da instanciação, por exemplo: `ComMaisMembros(1, 2, 3)` define `a`, `b` e `c`

Ao criar uma nova instância de `ComMaisMembros`, todos os quatro métodos de protótipo `setA`, `setB`, `setC` e `soma` estarão disponíveis.

Observe que se houver uma redefinição de uma propriedade ou método no novo tipo, a nova definição terá precedência.

Conversão de tipos

No FatScript, o símbolo `*` é usado para conversão de tipo, permitindo que você trate um tipo de dado como outro sem alterar os dados subjacentes. Essa capacidade é especialmente útil para especificar explicitamente o tipo ou para tratar valores como tipos compatíveis, por exemplo:

```
time.format(Epoch * 1688257765448) # trata o número como um valor de Época Unix
```

a conversão de tipos não altera a implementação subjacente, ela apenas rotula o valor com o tipo especificado, portanto, não pode ser usada para converter um número em texto ou em outros tipos incompatíveis

Aceitação flexível de tipos

FatScript oferece flexibilidade na aceitação de tipos por meio da inclusão de um tipo base. Este sistema permite a criação de tipos inter-relacionados que podem ser utilizados de forma intercambiável em métodos ou como elementos de uma Lista.

Por exemplo, consideremos os tipos `A`, `B`, e `C`. Se os tipos `B` e `C` incorporam o tipo `A` de maneira exclusiva em suas definições, eles são considerados como compartilhando as mesmas características derivadas de `A`, tornando `B` e `C` tipos compatíveis sob a base de `A`.

Aqui está como isso parece no código:

```
A = (_)
B = (A, b = true)
C = (A, c = true)

# method1 aceita ambos tipos B e C
method1 = (a: A) -> 'valid'

# essa lógica também se aplica a listas
mixedList: List/A = [ B(), C() ]
```

a flexibilidade do tipo só é possível se o tipo de dados é baseado em `Scope`

Advertência

Este sistema permite que um método projetado para aceitar um objeto do tipo `B` também possa aceitar um objeto do tipo `C` devido à base comum em `A`:

Tipos

```
method2 = (x: B) -> 'valid'  
method2(C()) # retorna 'valid' (inesperadamente?)
```

Embora o sistema flexível seja geralmente útil, ele pode ser inadequado quando é necessária uma correspondência exata de tipos. Nesses casos, poderia-se verificar explicitamente o tipo dentro do método, por exemplo, usando `x == B` para aceitar apenas objetos do tipo B.

Para restringir a flexibilidade de tipo e garantir correspondência exata, deve-se incluir `StrictType` na definição do tipo:

```
C = (A, StrictType, c = true) # C agora exige correspondência estrita de tipo
```

Esta modificação impede que C seja utilizado onde A ou B são aceitos, mesmo que ambos compartilhem do mesmo tipo base A.

Tipos compostos

No FatScript, tipos compostos permitem que você defina estruturas de dados complexas compostas por tipos mais simples para restringir aceitação de parâmetros em métodos e atribuições. Eles são representados usando barras / para separar os tipos na definição do tipo composto.

Vamos ver alguns exemplos e entender como os tipos compostos funcionam:

1. `ListOfNumbers = List/Number`, define um tipo composto `ListOfNumbers`, que é uma lista que só pode conter números.
2. `Matrix = List/List/Number`, define um tipo composto `Matrix`, que é uma lista de listas que só pode conter números.
3. `MethodReturningListOfNumbers = Method/ListOfNumbers`, define um tipo composto `MethodReturningListOfNumbers`, que é um método que retorna um `ListOfNumbers`.
4. `NumericScope = Scope/Number` define um tipo composto `NumericScope`, que é um escopo cujas entradas podem ser apenas do tipo número.

Veja também

- [Pacote type](#)

Any

Um tipo virtual que engloba todos os tipos e nenhum tipo ao mesmo tempo.

Tipo padrão

Any é o tipo inferido e o tipo de retorno quando nenhum tipo é explicitamente anotado em um método. Por exemplo:

```
identity = _ -> _
```

é equivalente a:

```
identity = (_: Any): Any -> _
```

Usar Any, seja implicitamente ou explicitamente, desabilita a verificação de tipos para um parâmetro. A anotação explícita pode ser útil em casos em que você deseja deixar claro que está dando flexibilidade ao tipo que é aceito.

Ser muito liberal com Any pode tornar seu código menos previsível e mais difícil de manter. É geralmente recomendado ser mais específico com as anotações de tipo sempre que possível:

Exemplo de uso de Any que pode levar a problemas

```
console <- fat.console
```

```
doubleIt = (arg: Any): Void -> console.log(arg * 2)
```

```
doubleIt(2)    # imprime: '4'
doubleIt('a') # gera: Error: unsupported expression > Text <multiply> Number
```

Este exemplo mostra que, embora a anotação de tipo Any permita flexibilidade no tipo do parâmetro, também pode resultar em comportamento inesperado se um argumento de um tipo inesperado for passado. Ao ser mais específico com a anotação de tipo, como Number, você pode tornar seu código mais previsível e autoexplicativo.

Exemplo de uso de uma anotação de tipo específica para maior previsibilidade

```
console <- fat.console
```

```
doubleIt = (num: Number): Void -> console.log(num * 2)
```

```
doubleIt(2)    # imprime: '4'
doubleIt('a') # gera: TypeError: type mismatch > num
```

Ao usar Number como a anotação de tipo, o método doubleIt agora é mais específico e só aceita argumentos do tipo Number.

Comparação

A única operação possível com Any é a comparação com ele, mas note que Any aceita todos os valores indistintamente, então não há uso prático para isso:

```
null      == Any # verdadeiro
true      == Any # verdadeiro
12345     == Any # verdadeiro
'abcd'    == Any # verdadeiro
[ 1, 2 ]  == Any # verdadeiro
{ a = 8 } == Any # verdadeiro
```

as comparações com Any não podem ser usadas para verificar a presença de um valor em um escopo, pois até mesmo null é aceito

Void

Quando você olha para o 'Vazio', apenas 'nulo' pode ser visto.

Tem alguém aí fora?

Uma entrada é avaliada como `null` se não estiver definida no escopo atual.

Você pode comparar com `null` usando igualdade `==` ou desigualdade `!=`, como:

```
a == null # verdadeiro, se 'a' não estiver definida
0 != null # verdadeiro, porque 0 é um valor definido
```

Tenha em mente que você não pode declarar uma entrada sem valor no FatScript.

Embora você possa atribuir `null` a uma entrada, isso causa comportamentos diferentes, dependendo se a entrada já existe no escopo e se é mutável ou não:

- Se uma entrada ainda não foi declarada, atribuir `null` a ela declara a entrada no escopo, mas a deixa sem um valor observável.
- Se uma entrada foi declarada e é `null`, atribuir `null` a ela não tem efeito.
- Se já existe, é não-nula e imutável, atribuir `null` gera um erro.
- Se já existe, é não-nula e mutável, atribuir `null` remove o valor.

Declaração de exclusão

Atribuir `null` a uma entrada mutável é o mesmo que excluir seu valor do escopo. Se excluído, seu tipo também é apagado.

```
~ m = 4 # entrada de número mutável
m = null # exclui m do escopo
```

entradas nulas são sempre mutáveis e podem transitar para um estado imutável quando um valor é atribuído

Comparações

Você também pode usar `Void` para verificar o valor de uma entrada, como:

```
() == Void # verdadeiro
null == Void # verdadeiro
false == Void # falso
0 == Void # falso
'' == Void # falso
[] == Void # falso
{} == Void # falso
```

Observe que `Void` só aceita `()` e `null`.

Formas de vazio

Em FatScript, o conceito de "vazio" ou a ausência de um valor pode ser representado de duas maneiras: usando `null` ou parênteses vazios `()`. Eles são efetivamente idênticos, em termos de comportamento no código:

```
null == null # verdadeiro
() == null # verdadeiro
() == () # verdadeiro
```

Usando null

A palavra-chave `null` denota explicitamente a ausência de um valor. É comumente usada em cenários onde um parâmetro ou valor de retorno pode não apontar para nenhum valor.

```
method(null, otherParam)
```

```
var = null
```

Void

Também pode ser usado para tornar um parâmetro opcional, permitindo que métodos sejam chamados com diferentes números de argumentos:

```
method = (mandatory: Text, optional: Text = null) -> {  
  ...  
}
```

`null` pode ser usado explicitamente em qualquer contexto onde uma ausência de valor precisa ser representada

Usando parênteses vazios

Quando usado no contexto de retornos de métodos, `()` pode significar que o método não retorna nenhum valor significativo.

```
fn = -> {  
  doSomething  
  
  ()  
}
```

Aqui, `fn` executa alguma ação e então usa `()` para indicar a ausência de um valor de retorno significativo, retornando efetivamente `void`.

a diferença reside no estilo de código, então isso é apenas uma sugestão, não uma regra rígida

Veja também

- [Extensões do protótipo Void](#)

Boolean

Booleanos são muito primitivos, eles só podem ser 'verdadeiro' ou 'falso'.

Comparação

Além de igualdade == e desigualdade !=, os booleanos também aceitam os seguintes operadores:

& AND lógico

```
true & true == true
true & false == false
false & true == false
false & false == false
```

AND interrompe a expressão se o lado esquerdo for falso

| OR lógico

```
true | true == true
true | false == true
false | true == true
false | false == false
```

OR interrompe a expressão se o lado esquerdo for verdadeiro

% XOR lógico (OR exclusivo)

```
true % true == false
true % false == true
false % true == true
false % false == false
```

XOR sempre avalia ambos os lados da expressão

Operador Bang

!! converte qualquer tipo em booleano, assim:

- null -> false
- zero (número) -> false
- não-zero (número) -> true
- vazio (texto/lista/escopo/chunk) -> false
- não-vazio (texto/lista/escopo/chunk) -> true
- método -> true
- erro -> false

AND/OR lógicos (&, |) e fluxos condicionais (==, ?) converterão implicitamente para booleano

Veja também

- [Extensões do protótipo Boolean](#)
- [Tipo Fuzzy](#)
- [Controle de fluxo](#)

Number

Um conceito matemático usado para contar, medir e fazer outras coisas de [matemáticas](#).

Declaração

O tipo `Number` é implementado como `double`. Veja como declarar um número:

```
a = 5           # declaração de número (imutável)
b: Number = 5  # mesmo efeito, com verificação de tipo
c: Number = a  # iniciando com o valor da entrada, também 5
d = 43.14     # com casas decimais
```

Para declarar uma entrada mutável, coloque o operador til antes:

```
~ a = 6 # entrada de número mutável
a += 1  # adiciona 1 a 'a', resultando em 7
```

Operações com números

Números aceitam várias operações:

- `==` igual
- `!=` diferente
- `+` soma
- `-` subtração
- `*` multiplicação
- `/` divisão
- `%` módulo
- `**` potência
- `<` menor
- `<=` menor ou igual
- `>` maior
- `>=` maior ou igual
- `&` AND lógico
- `|` OR lógico

Ressalvas

Para operações lógicas e controle de fluxo, lembre-se de que zero é considerado falso e um não-zero é considerado verdadeiro.

Para operadores de igualdade, embora `0` e `null` sejam avaliados como falsos, no FatScript eles não são iguais:

```
0 == null # falso
```

Precisão

Embora a precisão aritmética de um IEEE 754 `double` seja maior, o `fry` utiliza truques de arredondamento para melhorar a legibilidade humana ao imprimir sequências longas de decimais como texto. Além disso, ele usa um epsilon de `1.0e-06` para comparações de 'igualdade' entre números.

Em 99,999% dos casos de uso, essa abordagem fornece tanto comparações mais convenientes quanto números com aparência mais natural:

```
# Epsilon de igualdade
x = 1.0e-06
x: Number = 0.000001

# Diferenças menores são tratadas como o "mesmo" número pela comparação
x == 0.0000015
Boolean: true # a diferença de 0,0000005 é ignorada
```

Os números de ponto flutuante não são distribuídos uniformemente na reta numérica. Eles são densos em torno de 0 e, à medida que a magnitude aumenta, o 'delta' entre dois valores expressivos aumenta:

Text

Textos podem conter muitos caracteres e são às vezes chamados de strings.

Declaração

Entradas de texto são declaradas usando aspas:

```
a = 'hello world'      # declaração de texto inteligente
a = "hello world"     # declaração de texto bruto
a: Text = 'hello world' # inteligente, opcionalmente verboso
```

Manipulando texto

Concatenação

No FatScript, você pode concatenar, ou juntar, dois textos usando o operador `+`. Essa operação conecta os dois textos em um. Por exemplo:

```
x1 = 'ab' + 'cd' # Retorna 'abcd'
```

Subtração de texto

FatScript também suporta uma operação de subtração de texto usando o operador `-`. Essa operação remove uma substring especificada do texto. Por exemplo:

```
x2 = 'ab cd'
x2 - ' ' == 'abcd' # Retorna true
```

No exemplo acima, o caractere de espaço `' '` é removido do texto original `'ab cd'`, resultando em `'abcd'`.

Seleção de texto

A seleção permite que você acesse partes específicas de um texto usando índices. No FatScript, você pode usar índices positivos ou negativos. Os índices positivos começam do início do texto (`0` é o primeiro caractere), e os índices negativos começam do final do texto (`-1` é o último caractere).

para uma explicação detalhada sobre o sistema de indexação no FatScript, consulte a seção sobre acesso e seleção de itens em [List](#)

Quando apenas um índice é passado para a função de seleção, um único caractere do texto é selecionado. Quando um intervalo é passado para a função, um fragmento do texto é selecionado. Essa seleção é inclusiva, o que significa que inclui os caracteres nos índices inicial e final, a menos que se use o operador de intervalo semiaberto `..<`, exclusivo no lado direito.

Assim como com as listas, acessar itens que estão fora dos índices válidos irá gerar um erro. Para seleções, não são gerados erros ao acessar índices fora dos limites; em vez disso, um texto vazio é retornado.

```
x3 = 'exemplo'
x3(1)      # 'x'
x3(2, 4)   # 'emp'
x3(..2)    # 'exe'
x3(..<2)   # 'ex'
```

Manipulação direta

Mutação em textos permite alterar caracteres específicos por posição. Por exemplo, em `~ text = 'ai'`, você pode modificar o último caractere com `text[1] = 'e'`, transformando o texto em `'ae'`.

Caracteres especiais

Caracteres como aspas `' / "` podem ser escapados com a barra invertida `\`.

```
'Rock\n\roll'
"Onde fica \"aqui\"?"
```


você só precisa escapar as aspas do mesmo tipo usadas como delimitador de texto

Outras seqüências de escape suportadas são:

- backspace `\b`
- nova linha `\n`
- retorno de carro `\r`
- tabulação `\t`
- escape `\e`
- octeto em representação base-8 `\ooo`
- octeto em representação hexadecimal `\xhh`
- a própria barra invertida `\\`

Textos inteligentes

Quando declarado com aspas simples `'`, o modo inteligente é habilitado e a interpolação é realizada para qualquer código envolto em chaves `{...}`:

```
texto = 'mundo'
interpolado = 'olá {texto}' # resulta em 'olá mundo'
```

o template é processado em uma camada com acesso ao escopo atual

Observe que o uso de novas linhas ou outros textos inteligentes dentro do template de interpolação não é suportado, mas você pode fazer chamadas de método, se precisar compor o resultado com algo mais complexo.

Você pode evitar a interpolação escapando o colchete de abertura:

```
escapado = 'olá \{texto}' # resulta em 'olá {texto}'
```

Alternativamente, você pode evitar a interpolação usando textos brutos.

Textos brutos

Quando declarado com aspas duplas `"`, o modo de texto bruto é assumido e a interpolação é desativada.

Exemplo de modo inteligente vs. modo bruto:

```
'Sou inteligente: {interpolado}' # usando o valor do exemplo anterior
Sou inteligente: olá mundo      # substituição ocorre
```

```
"Sou bruto: {interpolado}" # colchetes são apenas caracteres comuns
Sou bruto: {interpolado}   # nenhuma interpolação ocorre
```

Operações com textos

- `==` igual
- `!=` diferente
- `+` soma (concatenar)
- `-` subtração (remove substring)
- `<` menor (alfanumérico)
- `<=` menor ou igual (alfanumérico)
- `>` maior (alfanumérico)
- `>=` maior ou igual (alfanumérico)
- `&` AND lógico (convertido para booleano)
- `|` OR lógico (convertido para booleano)

Codificação

FatScript é projetado para operar com textos codificados em UTF-8. Essa escolha de design reconhece a prevalência desses sistemas de codificação e otimiza a linguagem para ampla compatibilidade.

UTF-8 é um sistema de codificação de vários bytes capaz de representar qualquer caractere no padrão Unicode. Este esquema de codificação de caracteres universais usa de 8 a 32 bits para representar um caractere, permitindo a representação de uma vasta gama de símbolos de diversas línguas e sistemas de escrita. Notavelmente, os primeiros 128 caracteres (0-127) do UTF-8 se alinham precisamente com o conjunto ASCII, tornando qualquer texto ASCII uma string válida codificada em UTF-8.

Text

No FatScript, o tipo de dados `Text` é uma sequência de caracteres Unicode, inerentemente codificada em UTF-8, portanto, operações como `text.size`, `text(index)` e `text(1..4)` irão contar, acessar ou fatiar corretamente o texto, independentemente da complexidade dos caracteres. Essas operações consideram um caractere UTF-8 multi-byte completo como uma única unidade, garantindo um comportamento correto e previsível.

Veja também

- [Extensões do protótipo Text](#)

Method

Métodos são receitas que podem receber argumentos para "preencher as lacunas".

no FatScript, nos referimos a funções como Métodos, sem considerar o contexto de definição

Definição

Um método é definido anonimamente com uma seta fina `->`, assim:

```
<parâmetros> -> <receita>
```

Os parâmetros podem ser omitidos se nenhum for necessário:

```
-> <receita> # aridade zero
```

Para registrar um método no escopo, atribua-o a um identificador:

```
<identificador> = <parâmetros> -> <receita>
```

Parâmetros carregado para dentro do escopo de execução de um método são imutáveis, garantindo que as operações do método não alterem seu estado original. Para um comportamento mutável, considere passar um escopo ou utilizar um [tipo personalizado](#) capaz de encapsular múltiplos valores e estados.

Parâmetros opcionais

Enquanto assinaturas de métodos tipicamente requerem um número fixo de parâmetros obrigatórios, FatScript suporta parâmetros opcionais através de valores padrão:

```
greet = (nome: Texto = 'Mundo') -> {
  'Olá, {nome}'
}
```

```
greet() # 'Olá, Mundo'
```

Neste exemplo, o parâmetro `nome` é opcional, assumindo 'Mundo' como valor padrão se não for fornecido. Esta característica permite invocações de métodos mais flexíveis.

Argumento implícito

Uma conveniência oferecida pelo FatScript é a capacidade de fazer referência a um valor passado para o método sem precisar especificar um nome a ele explicitamente. Neste caso, o argumento implícito é então representado pelo sublinhado `_`.

Aqui está um exemplo que ilustra o uso do argumento implícito:

```
dobro = -> _ * 2
dobro(3) # saída: 6
```

Você pode usar um argumento implícito sempre que precisar realizar uma operação simples em um único parâmetro sem atribuir um nome específico a ele, mas note que o método deve ter aridade zero para ativá-lo.

Tratamento de argumentos

No FatScript, embora haja suporte de parâmetros opcionais e argumento implícito, quaisquer outros argumentos extras são simplesmente ignorados para melhorar a flexibilidade e o desempenho.

A decisão de design de ignorar argumentos extras também significa que não há suporte nativo para argumentos de comprimento variável no senso tradicional. Para obter funcionalidade semelhante, você pode declarar parâmetros opcionais tal como:

```
vaMethod = (v1 = null, v2 = null, v3 = null, v4 = null) -> ...
```

tenha em mente que você precisa listar explicitamente cada parâmetro que deseja capturar e definir um número muito grande de parâmetros (por exemplo, mais de 10) pode reduzir a performance das chamadas ao método

Auto-retorno

Method

FatScript usa o auto-retorno, ou seja, o último valor é retornado automaticamente:

```
resposta: Method = (aGrandeQuestao) -> {
  # TODO: explicar a vida, o universo e tudo mais
  42
}

resposta('6 x 7 = ?') # retorna: 42
```

Segurança do tipo de retorno

Em FatScript, uma peculiaridade é que, mesmo quando você declara um método com um tipo de retorno específico, a linguagem permite valores `null`, como em:

```
fn = (arg: Text): Text -> ... ? ... : null
```

Isso significa que, embora o método esteja declarado para retornar `Text`, o valor de retorno é, de certo modo, opcional, pois o método também pode retornar `Void`. A única garantia estrita é que, se o método tentar retornar um tipo incompatível, como `Number` ou `Boolean`, um `TypeError` será levantado. Essa escolha de design introduz uma flexibilidade implícita, mantendo, ao mesmo tempo, um certo nível de segurança de tipos.

Se você precisa garantir um resultado não nulo, pode envolver sua chamada com [Option](#) assim:

```
Option(fn(myArg)).getOrElse('fallbackVal')
```

Procedimentos

O FatScript introduz um recurso único que simplifica as chamadas de métodos quando não envolvem argumentos.

O símbolo `<>` declara um procedimento (Procedure), uma função sem argumentos que executa automaticamente quando referenciada:

```
<identificador> = <tipo> <> <receita>
```

para a sintaxe de procedimento, o `tipo` precisa ser uma única palavra; se você precisar de um [tipo composto](#), declare-o como um alias de antemão e use o alias

passar argumentos para um procedimento resultará em erro, pois procedimentos não aceitam argumentos

Principais vantagens:

- **Redução de código:** Reduz a necessidade de parênteses, tornando o código mais limpo e conciso, para procedimentos sem parâmetros que agem como propriedades.
- **Computação dinâmica:** Permite a computação dinâmica com saídas que podem mudar com base no estado interno do objeto ou no estado global.
- **Execução diferida:** Habilita a execução diferida, útil em programação assíncrona e padrões de inicialização complexos.

a partir da versão 4.0.0, somente procedimentos suportam a execução automática sem parênteses; métodos clássicos de aridade zero não são mais executados automaticamente e exigem parênteses `()` para a execução

Evitando uma chamada automática

Para referenciar um procedimento sem acionar o recurso de chamada automática, você pode usar a sintaxe de obtenção:

```
foo('bar') # gera uma referência para foo.bar, sem chamá-lo
```

O FatScript também oferece as palavras-chave `self` e `root` para referenciar procedimentos nos níveis local e global, respectivamente:

```
self('meuProcedimentoLocal')
root('meuProcedimentoGlobal')
```

O operador til `~` também permite que você contorne o recurso de chamada automática, proporcionando flexibilidade no manuseio de procedimentos:

```
# Ambas as linhas abaixo buscam a referência do procedimento, sem chamá-lo
foo.~bar
~ meuProcedimento
```

Rótulos de argumento

O FatScript suporta rótulos de argumento, que permitem que você especifique nomes para argumentos no local da chamada. Esses rótulos melhoram a legibilidade do código e a autodocumentação ao tornar a intenção de cada argumento explícita:

```
# Definindo um método com parâmetros
fn = (a: Número, b: Número) -> a + b

# Chamando o método com rótulos de argumento
fn(a = 1, b = 2) # saída: 3, o mesmo que fn(1, 2)
```

Se fornecidos, os rótulos são validados em relação aos nomes de parâmetros do método. Os argumentos devem ser passados na mesma ordem definida na assinatura do método. Usar rótulos incorretos gerará um erro:

```
fn(b = 1, a = 2) # CallError: invalid name 'b' at pos: 1
```

argumentos são resolvidos **seqüencialmente**, não pelos rótulos; portanto, **resolução fora de ordem não é permitida**, mesmo quando rótulos são usados

Contraste com a instanciação de tipo

Embora rótulos de argumento em chamadas de método sejam praticamente decorativos, eles desempenham um **papel funcional** na [instanciação de tipo](#). Ao criar instâncias de tipos, rótulos de argumento são correspondidos por nome às propriedades do tipo, permitindo **resolução fora de ordem**.

Ao manter a resolução seqüencial para métodos, o FatScript garante melhor desempenho em chamadas de método, enquanto a instanciação de tipo se beneficia da flexibilidade da resolução de argumento nomeado.

Otimização de Recursão de Cauda

O FatScript suporta Otimização de Recursão de Cauda (TRO, do inglês *Tail Recursion Optimization*) para melhorar o desempenho, conservando espaço na pilha. Para se beneficiar desta otimização, várias condições devem ser atendidas:

1. **Parâmetros explícitos:** Os métodos devem declarar explicitamente os parâmetros; a funcionalidade de argumento implícito não é suportada para TRO.
2. **Controle de fluxo:** TRO é compatível apenas com as construções `If-Else`, `Cases` e `Switch` para ramificação.
3. **Estrutura de chamada:** Chamadas de métodos aninhadas, como `x(a)(b)(c)`, não são suportadas para TRO.
4. **Chamadas recursivas:** O método deve se chamar recursivamente pelo nome como a operação final em seu caminho de execução.

Por exemplo, uma função configurada corretamente para TRO pode parecer assim:

```
tailRec = (n: Number, m: Number): Void -> {
  n > m => console.log('concluído')
  - => {
    console.log(n)
    tailRec(n + 1, m)
  }
}
```

Neste exemplo, `tailRec` se chama recursivamente como a operação final em um dos ramos, tornando-a elegível para otimização.

Você pode verificar se TRO foi ativada para o seu método usando [análise estática](#) com a opção `fry --probe`.

A TRO pode ser desativada envolvendo a chamada recursiva entre parênteses, conforme mostrado abaixo:

```
...
(tailRec(n + 1, m)) # sem TRO
```

Veja também

Method

- [Extensões do protótipo Method](#)

List

Listas são coleções ordenadas de itens do mesmo tipo, acessados por índice.

Definição

As listas são definidas com colchetes [], como no exemplo a seguir:

```
lista: List/Text = [ 'maçã', 'pizza', 'pêra' ]
```

Listas não permitem a mistura de tipos. O tipo de uma lista é determinado pelo primeiro item adicionado a ela; conseqüentemente, as listas vazias não têm tipo.

As listas pulam posições vazias, então um item que avalia para null é ignorado:

```
a = 1
c = 3
[ a, b, c ] # retorna: [ 1, 3 ] (b é ignorado)
```

Acesso

Itens individuais

Os itens da lista podem ser acessados individualmente com chamada de índice baseado em zero:

```
lista(0) # 'maçã'
lista(2) # 'pêra'
```

Valores negativos indexam a lista de trás para frente, começando em -1 como o último item:

```
lista(-1) # 'pêra'
```

O acesso a itens que estão fora dos índices válidos gera um erro:

```

      0      1      2      > 2
Erro [ 'maçã', 'pizza', 'pêra' ] Erro
    < -3      -3      -2      -1
```

Seleções

Os índices de início e fim funcionam exatamente da mesma maneira que ao acessar itens individuais; então, os valores negativos contam a partir do último item e podem ser regressivos. No entanto, ao usar intervalos, nenhum erro é gerado ao acessar índices fora dos limites; em vez disso, uma lista vazia é retornada.

```
lista(0..0) # [ 'maçã' ]
lista(4..8) # []
lista(1..-1) # [ 'pizza', 'pêra' ]
```

Um índice pode ser deixado em branco, e o início a partir do primeiro item ou o fim no último item é assumido:

```
lista(..1) # [ 'maçã', 'pizza' ]
lista(1..) # [ 'pizza', 'pêra' ]
```

Listas aninhadas

Uma matriz pode ser usada e acessada da seguinte maneira:

```
matriz = [
  [ 1, 2, 3 ]
  [ 4, 5, 6 ]
]
```

```
matriz(1)(0) # retorna 4 (1: segunda linha, em seguida, 0: primeiro índice)
```

para simplificar, o exemplo usa uma matriz 2D, mas poderia ser n-dimensional

Operações

- == igual
- != diferente
- + adição (efeito de concatenação)
- - subtração (efeito de diferença)
- & AND lógico
- | OR lógico

AND/OR lógicos avaliam listas vazias como `false`, caso contrário `true`

Adição de lista (concatenação)

A operação de adição de listas permite combinar duas listas em uma nova lista:

```
x = [ 1, 2, 2, 3 ]
y = [ 3, 3, 4, 4 ]
```

```
x + y # resultado: [ 1, 2, 2, 3, 3, 3, 4, 4 ]
```

Nesse caso, ao usar o operador de adição `+` para unir as listas `x` e `y`, os elementos de ambas as listas são combinados em uma única lista. A ordem dos elementos na lista resultante é determinada pela ordem em que as listas foram adicionadas.

não há remoção de elementos duplicados durante a concatenação

Concatenação rápida

Para melhor desempenho, você pode aproveitar o operador `+=`, por exemplo:

```
~ lista += [ valor ] # mais rápido
```

tem mesmo efeito que alternativas mais lentas:

```
~ lista = []
lista = lista + [ valor ] # concatenação explícita
lista[lista.size] = valor # indexado por 'size', membro do protótipo
```

Outro detalhe do operador `+=`, que se aplica também a outros tipos, é a inicialização automática por omissão, onde caso a entrada ainda não tenha sido declarada anteriormente, atua como uma simples atribuição.

Subtração de lista (diferença)

A operação de subtração de listas, permite remover os elementos do segundo operando que estão presentes no primeiro operando, resultando em uma lista contendo apenas valores únicos:

```
x = [ 1, 2, 2, 3 ]
y = [ 3, 3, 4, 4 ]
```

```
x - y # resultado: [ 1, 2 ]
y - x # resultado: [ 4 ]
```

Nesse caso, ao subtrairmos a lista `y` da lista `x`, os elementos com valor 3 são removidos, já que estão presentes em ambas as listas. O resultado é a lista `[1, 2]`. Da mesma forma, ao subtrairmos a lista `x` da lista `y`, o único elemento restante é o valor 4.

apenas valores exatamente idênticos são removidos durante a subtração

Veja também

- [Extensões do protótipo List](#)
- [Mapeando uma lista](#)

Scope

Um escopo é semelhante a um dicionário, onde valores estão associados a chaves.

Definição

Escopos são definidos usando chaves {}, conforme mostrado abaixo:

```
meuEscopoBacana = {
  lugar = 'aqui'
  quando = 'agora'
}
```

Escopos armazenam entradas em ordem alfabética, uma característica que se torna aparente ao [mapear sobre um escopo](#).

Acesso

Há três maneiras de acessar diretamente as entradas dentro de um escopo.

Sintaxe de ponto

```
meuEscopoBacana.lugar # retorna: 'aqui'
```

Sintaxe de obtenção

```
# assumindo que prop = 'lugar'
meuEscopoBacana(prop) # retorna: 'aqui'
```

Em ambos os métodos, se a propriedade não estiver presente, `null` é retornado. Se o escopo externo não for encontrado, um erro é gerado.

Sintaxe de encadeamento opcional

Use o operador interrogação-ponto `?.` para encadear com segurança escopos externos potencialmente inexistentes:

```
naoExistente?.prop # retorna null
```

A sintaxe de encadeamento opcional não gera um erro quando o escopo externo é `null`.

Operações

- `==` igual
- `!=` diferente
- `+` adição (efeito de mesclagem)
- `-` subtração (efeito de diferença)
- `&` AND lógico
- `|` OR lógico

AND/OR lógico avaliam escopos vazios como `false`, caso contrário, `true`

Adição de escopo (mesclagem)

O segundo operando age como um patch para o primeiro:

```
x = { a = 1, b = 3 }
y = { b = 2 }
```

```
x + y # resulta em { a = 1, b = 2 }
y + x # resulta em { a = 1, b = 3 }
```

valores do segundo operando substituem os do primeiro

Subtração de escopo (diferença)

A subtração remove elementos do primeiro operando que são idênticos aos do segundo:

Scope

```
x = { a = 1, b = 3 }
y = { a = 1 }

x - y # resulta em { b = 3 }
```

apenas valores exatamente idênticos são removidos

Blocos Escopados

Blocos Escopados em FatScript permitem a execução de declarações dentro do contexto de um escopo específico:

```
objeto.{
  # Declarações executadas no contexto de 'objeto'
}
```

Aqui, `objeto` é o escopo alvo. Dentro do bloco, você pode acessar e modificar diretamente as propriedades de `objeto`.

Características

- **Isolamento:** entradas declaradas dentro de um Bloco Escopado são locais para aquele escopo e não afetam o escopo externo.
- **Acesso ao Escopo Externo:** Blocos Escopados podem acessar entradas do escopo externo.

Exemplo

```
x = {}

x.{
  a = 5      # 'a' agora é uma propriedade de 'x'
  b = a + 3  # 'b' agora também é uma propriedade de 'x'
}
```

Interações de escopos

FatScript utiliza mecanismos sofisticados para gerenciar variáveis em diferentes escopos, aproveitando conceitos de escopo léxico e sombreamento para fornecer capacidades de programação poderosas. Esta seção explora esses mecanismos, incluindo nuances de atribuição, comportamentos de incremento/decremento e o uso inovador do operador `+=` para alternância de booleanos.

Atribuição

O operador de atribuição (`=`) copia valores de escopos externos para o escopo atual, definindo um novo valor:

```
~ n = 1
x = {}
x.{ ~ n = n } # agora x.n == 1, e x.n é independente de root.n
x.{ c = n }   # tem efeito semelhante, porém 'c' é imutável
```

o mesmo conceito se aplica ao código executado em um escopo de método

Ressalva

Usar `~ n = n + 1` dentro de um bloco ou método adiciona um novo 'n' no escopo atual, inicializado com o valor de `n + 1` do escopo envolvente mais próximo, sem alterar o 'n' externo.

Incrementando e decrementando

Operações de incremento (`+=`) e decremento (`-=`) interagem com o escopo de variáveis de uma maneira diferente. Essas operações buscam a instância mais próxima de uma variável, começando do escopo atual e movendo-se para fora recursivamente, e então modificam essa instância diretamente.

```
~ outerN = 1
fn = -> {
  outerN += 1 # alveja e incrementa 'outerN' no escopo externo
}
```

Auto-inicialização com `+=`

Scope

FatScript também fornece um comportamento especial em relação ao operador de incremento (`+=`). Se não existir o incremento funciona como uma atribuição regular como se você tivesse escrito o seguinte para `n += 1`:

```
n == Void ? n = 1 : n += 1
```

O recurso de auto-inicialização pode ser particularmente útil quando usado em combinação com [entradas dinâmicas](#) para programação dinâmica.

este recurso está disponível exclusivamente para o operador de incremento, decremento não pode inicializar valores não existentes

Alternância de booleanos com `+=`

Geralmente, booleanos não permitem operações de adição. FatScript, no entanto, estende a funcionalidade do operador `+=` para tipos booleanos, permitindo um mecanismo de alternância intuitivo dentro de escopos internos.

A expressão `flag += !flag` alterna efetivamente o valor booleano, mesmo quando `flag` é definido em um escopo externo.

no caso particular de booleanos, a única distinção entre `=` e `+=` é o escopo

Outros operadores de atribuição compostos

Da mesma forma, outras operações de atribuição compostas, como `*=`, `/=`, `%=` e `**=`, são suportados por tipos numéricos e respeitam as mesmas regras de escopo que se aplicam às operações de incremento e decremento.

Veja também

- [Entradas dinâmicas](#)
- [Extensões do protótipo Scope](#)
- [Mapeando um escopo](#)

Error

Há grande sabedoria em esperar pelo inesperado também.

Subtipos padrão

Embora alguns erros possam ser gerados com o tipo base Error, a maioria é [subtipada](#).

Veja as definições nas [extensões do protótipo Error](#).

Declaração

Erros também podem ser gerados explicitamente; você deve usar o [construtor de tipo](#):

```
_ <- fat.type.Error
Error('ocorreu um erro') # gera um erro genérico
MeuErro = Error
MeuErro('ocorreu outro erro') # gera um erro do subtipo MeuErro
```

Comparações

Erros sempre avaliam como falso:

```
Error() ? 'é verdadeiro' : 'é falso' # é falso
```

Erros são comparáveis ao seu tipo:

```
Error() == Error # verdadeiro
```

leia também a sintaxe de [comparação de tipo](#)

Uma maneira ingênua de lidar com erros poderia ser:

```
_ <- fat.console
# lidando com o erro retornado
talvezFalhe() <= Error => log('um erro aconteceu')
_ => log('sucesso')
```

isso só funciona se a [opção](#) -e / continuar em caso de erro estiver definida

Uma outra maneira ingênua de lidar com errors, mas que funciona sempre é utilizar uma [operação padrão](#):

```
talvezFalhe() ?? log('um erro aconteceu')
```

Embora a abordagem ingênua possa funcionar, uma maneira mais adequada de lidar com erros é definindo um manipulador de erro usando o método `trapWith` encontrado na [biblioteca failure](#).

Veja também

- [Biblioteca failure](#)
- [Extensões do protótipo Error](#)

Chunk

Chunks são apenas blocos binários de dados.

Declaração

Chunks não podem ser declarados explicitamente; você deve usar o [construtor de tipo](#) e aplicar uma das seguintes estratégias:

```
_ <- fat.type.Chunk
```

```
Chunk(null)           # Void -> [] (chunk vazio)
Chunk(4)              # Number -> [ 0, 0, 0, 0 ] (bytes em branco)
Chunk('ABC')          # Text -> [ 65, 66, 67 ]
Chunk([ 65, 66, 67 ]) # List/Number -> [ 65, 66, 67 ]
```

espera-se que lista de números contenha valores de byte válidos (0-255), senão um erro é gerado

Manipulando chunks

Concatenação

No FatScript, você pode concatenar, ou juntar, dois chunks usando o operador +. Por exemplo:

```
abCombinados = chunkA + chunkB
```

Seleção de blocos

A seleção permite acesso a partes específicas de um chunk usando índices. O FatScript suporta índices positivos e negativos. Índices positivos começam a partir do início do chunk (com 0 como o primeiro byte), enquanto índices negativos começam a partir do final (-1 é o último byte).

para uma explicação detalhada sobre o sistema de indexação no FatScript, consulte a seção sobre acesso e seleção de itens em [Lista](#)

Selecionar com um índice recupera um único byte do chunk (como número). Usando um intervalo de bytes, seleciona um fragmento incluindo os índices de início e fim, exceto quando se usa o operador de intervalo semiaberto ..<, que é exclusivo no lado direito.

Acessar índices fora do intervalo válido gerará um erro para seleções individuais. Para seleções de intervalo, índices fora dos limites resultam em um chunk vazio.

```
x3 = Chunk('example')
x3(1)   # 120 (valor ASCII de 'x')
x3(..2) # novo Chunk contendo 3 bytes (correspondendo a 'exa')
```

Manipulação direta

Mutação em dados binários permite alterar bytes específicos por posição.

Por exemplo, em ~ chunk = Chunk([65, 66, 67]), você pode modificar o último byte com chunk[2] = 68, transformando o dado em [65, 66, 68].

Comparações

São suportadas comparações de igualdade == e desigualdade != de chunks.

Veja também

- [Extensões do protótipo Chunk](#)

Controle de fluxo

Avance em um fluxo contínuo de decisões que devem ser tomadas.

Fallback

As operações padrão ou de coalescência nula são definidas com dois pontos de interrogação ?? e funcionam da seguinte maneira:

```
<talvezNullouError> ?? <valorFallback>
```

Caso o lado esquerdo não seja null nem Error, então ele é usado; caso contrário, fallbackValue é retornado.

de forma semelhante, você pode usar o operador de atribuição por coalescência nula ??=

If

Declarações If são definidas com um ponto de interrogação ?, como abaixo:

```
<condição> ? <resposta>
```

como não há alternativa, null é retornado se a condição não for atendida

If-Else

Declarações If-Else são definidas com um ponto de interrogação ? seguido de dois pontos :, como abaixo:

```
<condição> ? <resposta> : <alternativa>
```

Para usar declarações If-Else multilinhas, envolva a resposta em chaves {...} assim:

```
<condição> ? {
  <resposta>
} : {
  <alternativa>
}
```

Cases

Cases são definidos com a seta espessa => e são automaticamente encadeados, criando uma sintaxe intuitiva e simplificada, semelhante a uma declaração switch, sem a possibilidade de queda (fall-through). Isso permite que condições não relacionadas sejam misturadas, resultando em uma estrutura "if-else-if-else" mais concisa:

```
<condição01> => <respostaPara1>
<condição02> => <respostaPara2>
<condição03> => <respostaPara3>
...
```

Exemplo:

```
escolha = (x) -> {
  x == 1 => 'a'
  x == 2 => 'b'
  x == 3 => 'c'
}
```

```
escolha(2) # 'b'
escolha(8) # null
```

Para fornecer um valor padrão para seu método, você pode adicionar um caso pega-tudo usando um sublinhado _ no final da sequência:

```
escolha = (x) -> {
  x == 1 => 'a'
  x == 2 => 'b'
  x == 3 => 'c'
  _ => null
}
```

```

-      => 'd'
}

escolha(2) # 'b'
escolha(8) # 'd'

```

Para cenários mais complexos, você pode usar blocos como resultados para cada caso:

```

...
condição => {
  # faça algo
  'foo'
}
- => {
  # faça outra coisa
  'bar'
}
...

```

Cases devem terminar em um caso pega-tudo `_` ou final do bloco. O uso mais efetivo de Cases é dentro de métodos na parte inferior do corpo do método.

Embora seja possível adicionar Cases aninhados, é melhor evitar construções excessivamente complexas. Isso torna o código mais difícil de seguir e provavelmente perde o objetivo de usar esse recurso.

Pode ser mais apropriado extrair essa lógica para um método separado. O FatScript incentiva os desenvolvedores a dividir a lógica em métodos distintos, ajudando a evitar código spaghetti.

Switch

O operador Switch é denotado pelo símbolo de seta dupla para a direita `>>`, que guia o fluxo de controle baseado na correspondência do valor com uma série de casos:

Sintaxe:

```

<valor> >> {
  <valorCaso1> => <respostaPara1>
  <valorCaso2> => <respostaPara2>
  ...
  _ => <respostaPadrão>
}

```

Cada caso no bloco Switch é avaliado em ordem até que uma correspondência seja encontrada e o resultado do caso correspondente é retornado:

```

escolher = -> _ >> {
  1 => 'um'
  2 => 'dois'
  3 => 'três'
  _ => 'outro'
}

escolher(2) # 'dois'
escolher(4) # 'outro'

```

Os casos do Switch também podem envolver expressões, permitindo correspondências dinâmicas:

```

avaliar = (x, y) -> x >> {
  y + 1 => 'logo acima de y'
  y - 1 => 'logo abaixo de y'
  _     => 'não diretamente ao redor de y'
}

avaliar(5, 4) # 'logo acima de y'
avaliar(3, 4) # 'logo abaixo de y'
avaliar(7, 4) # 'não diretamente ao redor de y'

```

Tap

O operador `Tap` é representado pelo símbolo de duas setas para a esquerda `<<`, que facilita a execução de efeitos colaterais sem alterar o resultado principal de uma expressão. Ele é projetado para processar valores por meio de métodos especificados (taps) que podem executar efeitos colaterais, mas ainda retornando o valor original da expressão.

Sintaxe:

```
<resultado> << <métodoTap>
```

o lado direito de um `tap` deve sempre ser um [método](#)

Nesta estrutura, `<resultado>` é uma expressão cujo valor é passado para `<métodoTap>`, que executa usando `<resultado>` como sua entrada, mas não afeta o valor final da expressão. Em vez disso, `<métodoTap>` é usado puramente por seus efeitos colaterais.

Veja como o operador `tap` pode ser usado:

```
console <- fat.console  
increment = x -> x + 1  
resultado = increment(4) << console.log
```

Neste exemplo, `increment(4)` calcula para 5, que é então passado para `console.log` e, embora `console.log` retorne `null`, o resultado final atribuído a `resultado` é 5.

Vários efeitos colaterais podem ser encadeados sequencialmente, cada um recebendo o mesmo resultado inicial:

```
val = pure(in) << fx1 << fx2 << fx3
```


Loops

Repetir, repetir, repetir, repetir, repetir...

Sintaxe base

Todos os loops são criados com o sinal de arroba @, por exemplo:

```
<expressão> @ <corpoDoLoop>
```

Loop tipo "enquanto"

O corpo do loop irá executar enquanto a expressão avaliar para:

- verdadeiro
- número não-zero
- texto/chunk não-vazio

A execução irá terminar quando a expressão avaliar para:

- falso
- nulo
- número zero
- texto/chunk vazio
- erro

Por exemplo, este loop imprime números de 0 a 3:

```
_ <- fat.console
~ i = 0
(i < 4) @ {
  log(i)
  i += 1
}
```

Sintaxe de mapeamento

Você pode mapear intervalos, listas e escopos com um mapeador, assim:

```
<intervalo|coleção> @ <mapeador>
```

Uma nova lista é gerada com base nos valores de retorno do mapeador.

Mapeando um intervalo

Utilizando o operador de intervalo `..` o mapeador receberá um número como entrada sequencialmente do limite esquerdo até o limite direito:

```
4..0 @ num -> num + 1 # retorna [ 5, 4, 3, 2, 1 ]
```

a sintaxe de intervalo é inclusiva em ambos os lados, por exemplo, `0..2` retorna 0, 1, 2

Há também o operador de intervalo semiaberto `..<`, exclusivo no lado direito.

ressalva: o intervalo semiaberto não funciona com direção inversa, sempre precisa ser do mínimo para máximo

Mapeando uma lista

O mapeador receberá os itens em ordem (da esquerda para a direita):

```
[ 3, 1, 2 ] @ item -> item + 1 # retorna [ 4, 2, 3 ]
```

Mapeando um escopo

Loops

O mapeador receberá os nomes (chaves) das entradas armazenadas no escopo em ordem alfabética:

```
{ c = 3, a = 1, b = 2 } @ chave -> chave # retorna [ 'a', 'b', 'c' ]
```

nos exemplos, usamos literais de lista e escopo, mas uma entrada ou chamada que avalia para uma lista ou um escopo terá o mesmo efeito

Você pode acessar as entradas de um escopo referindo-se a ele pelo nome, mas neste caso precisa que ele esteja definido no escopo externo, por exemplo:

```
meuEscopo = { c = 3, a = 1, b = 2 }  
meuEscopo @ chave -> meuEscopo(chave) # retorna [ 1, 2, 3 ]
```

O FatScript utiliza um recurso de caching inteligente que faz com que esta sintaxe não gere um esforço adicional para buscar o elemento da vez no escopo durante o mapeamento.

Bibliotecas

Vamos falar sobre os doces recheios embutidos no FatScript: as bibliotecas!

Bibliotecas padrão

Essenciais

Estas são as bibliotecas fundamentais que você espera que estejam disponíveis em uma linguagem de programação, fornecendo funcionalidades essenciais:

- [async](#) - Trabalhadores e tarefas assíncronas
- [bridge](#) - Ponte entre FatScript e bibliotecas C externas
- [color](#) - Códigos de cores ANSI para console
- [console](#) - Operações de entrada e saída do console
- [curses](#) - Interface de usuário baseada em terminal
- [enigma](#) - Métodos de criptografia, hash e UUID
- [failure](#) - Tratamento de erros e gerenciamento de exceções
- [file](#) - Operações de entrada e saída de arquivos
- [http](#) - Framework de manipulação HTTP
- [math](#) - Operações e funções matemáticas
- [recode](#) - Conversão de dados entre vários formatos
- [sdk](#) - Utilitários do kit de desenvolvimento de software do fry
- [smtp](#) - Framework de manipulação SMTP
- [system](#) - Operações e informações no nível do sistema
- [time](#) - Manipulação de data e hora

Pacote de tipos

[Este pacote](#) estende os recursos dos [tipos nativos](#) do FatScript:

- [Void](#)
- [Boolean](#)
- [Number](#)
- [HugeInt](#)
- [Text](#)
- [Method](#)
- [List](#)
- [Scope](#)
- [Error](#)
- [Chunk](#)

Pacote Extra

[Tipos adicionais](#) implementados em FatScript puro:

- [Date](#) - Gerenciamento de calendário e datas
- [Duration](#) - Construtor de duração em milissegundos
- [Fuzzy](#) - Valores probabilísticos e operações de lógica fuzzy
- [HashMap](#) - Armazenamento rápido de chave-valor
- [Logger](#) - Suporte ao registro de logs
- [Memo](#) - Utilitário de memoização genérica
- [MouseEvent](#) - Parser de eventos de mouse
- [Opaque](#) - Utilitário para proteção suave de dados
- [Option](#) - Encapsulamento de valor opcional
- [Param](#) - Verificação de presença e tipo de parâmetro
- [Sound](#) - Interface de reprodução de som
- [Storable](#) - Armazenamento de dados

Atalho de importação

Se você deseja torná-los todos disponíveis de uma vez, pode simplesmente fazer o seguinte, e todas essas coisas boas estarão disponíveis para o seu código:

Bibliotecas

```
_ <- fat._
```

Embora esse recurso possa ser conveniente ao experimentar no REPL, esteja ciente de que ele traz todas as constantes e nomes de método da biblioteca, potencialmente poluindo seu namespace global.

fat.std

Alternativamente, importe a biblioteca "standard" que importa todos os tipos (inclusive do pacote extra), bem como aplica importações nomeadas de todos os demais pacotes, assim:

```
_ <- fat.std
```

O que equivale a:

```
_ <- fat.type._
_ <- fat.extra._
async <- fat.async
bridge <- fat.bridge
color <- fat.color
console <- fat.console
curses <- fat.curses
enigma <- fat.enigma
failure <- fat.failure
http <- fat.http
file <- fat.file
math <- fat.math
recode <- fat.recode
sdk <- fat.sdk
smtp <- fat.smtp
system <- fat.system
time <- fat.time
```

Observe que, a importação de tudo antecipadamente pode adicionar uma sobrecarga desnecessária ao tempo de inicialização do seu programa, mesmo que você precise usar apenas alguns métodos.

Como boa prática, considere importar apenas os módulos específicos de que você precisa, com [importações nomeadas](#). Dessa forma, você pode manter seu código limpo e conciso, minimizando o risco de conflitos de nome ou problemas de desempenho.

Hacking e mais

Sob o capô, as bibliotecas são construídas usando comandos embutidos. Para obter uma compreensão mais profunda e explorar o funcionamento interno do interpretador, mergulhe [neste tópico mais avançado](#).

async

Trabalhadores e tarefas assíncronas

Importação

```
_ <- fat.async
```

Tipos

A biblioteca `async` introduz o tipo `Worker`.

Worker

O `Worker` é um simples invólucro para uma operação assíncrona.

Construtor

Nome	Assinatura	Breve descrição
<code>Worker</code>	<code>(task: Method)</code>	Cria um <code>Worker</code> em modo espera

O construtor `Worker` aceita dois argumentos:

- **task**: O método a ser executado de forma assíncrona (o método pode não aceitar argumentos diretamente, mas você pode adicionar esses usando duas setas na definição `-> ->`).
- **wait** (opcional): O tempo limite em milissegundos. Se a tarefa não terminar dentro desse tempo, ela será cancelada.

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>start</code>	<code><> Worker</code>	Inicia a tarefa
<code>cancel</code>	<code><> Void</code>	Cancela a tarefa
<code>await</code>	<code><> Worker</code>	Aguarda pela conclusão da tarefa
<code>isDone</code>	<code><> Boolean</code>	Verifica se a tarefa foi concluída
<code>hasStarted</code>	<code>Boolean</code>	Definido pelo método <code>start</code>
<code>hasAwaited</code>	<code>Boolean</code>	Definido pelo método <code>await</code>
<code>isCanceled</code>	<code>Boolean</code>	Definido pelo método <code>cancel</code>
<code>result</code>	<code>Any</code>	Definido pelo método <code>await</code>

Métodos avulsos

Nome	Assinatura	Breve descrição
<code>atomic</code>	<code>(op: Method): Any</code>	Executa a operação atomicamente
<code>selfCancel</code>	<code><> Void</code>	Encerra a execução da thread
<code>processors</code>	<code><> Number</code>	Retorna o número de processadores

Notas de uso

As instâncias de `Worker` são mapeadas para threads do sistema em uma base um-para-um e são executadas conforme o agendamento do sistema. Isso implica que sua execução pode nem sempre ser imediata. Para aguardar o resultado de um `Worker`, use o método `await`.

Diferentemente de outros contextos, no código assíncrono, a `task: Method` executa sem acesso ao escopo no qual é criada. Ela só pode acessar propriedades que foram 'curryficadas' `-> ->` para dentro do seu escopo de execução ou aquelas que estão diretamente acessíveis no escopo global.

O limite global de memória é compartilhado por todos os `Workers`, mas um contexto completamente novo, incluindo uma pilha separada, é alocado para cada um. No entanto, em caso de um erro irreversível ou fatal, como a exaustão de memória ou

async

de pilha por um dos Workers, o interpretador será interrompido e todas as threads serão encerradas.

para manter o máximo desempenho, evite usar [interpolação de texto](#) dentro de tarefas assíncronas

Exemplos

```
async <- fat.async
math <- fat.math
time <- fat.time

# Define uma tarefa lenta
slowTask = (seconds: Number): Text -> -> {
  time.wait(seconds * 1000)
  'done'
}

# Inicia a tarefa como Worker
worker = Worker(slowTask(5)).start

# Obtém o resultado do worker
result1 = worker.await.result # bloqueia até a tarefa ser concluída

# Inicia uma tarefa com timeout
task = Worker(slowTask(5), 3000).start # a tarefa deve expirar

# Obtém o resultado da tarefa
result2 = task.await.result # bloqueia até a tarefa ser concluída ou ocorrer timeout

o método await da gera um AsyncError se a tarefa exceder o tempo antes da conclusão
```

atomic

O encapsulador `atomic` é uma ferramenta crítica para garantir a segurança das threads e a integridade dos dados na programação concorrente. Quando múltiplos workers ou tarefas assíncronas acessam e modificam recursos compartilhados, condições de corrida podem ocorrer, levando a resultados imprevisíveis e errôneos. A operação `atomic` aborda essa questão garantindo que o método que ela envolve seja executado de forma atômica. Isso significa que a operação inteira é completada como uma única unidade indivisível, sem possibilidade de outras threads intervirem no meio do caminho para a mesma operação. Isso é particularmente importante para operações como incrementar um contador, atualizar estruturas de dados compartilhadas ou arquivos, ou realizar qualquer ação onde a ordem de execução importa:

```
async.atomic(-> file.append(logFile, line))
```

Embora as operações `atomic` sejam uma ferramenta poderosa para garantir a consistência, é importante estar atento ao potencial de contenção que elas introduzem. A contenção ocorre quando múltiplas "threads" ou tarefas tentam executar uma operação simultaneamente, levando a potenciais gargalos de desempenho à medida que cada tarefa espera sua vez. O uso excessivo ou desnecessário de operações `atomic` pode degradar significativamente o desempenho de sua aplicação, reduzindo a concorrência. Mantenha apenas a seção crítica de código que absolutamente requer atomicidade envolvida como uma operação `atomic`.

por baixo dos panos, operações atômicas são fundamentalmente protegidas por um único mutex global

Async na Web Build

Ao usar `fry` compilado com Emscripten (por exemplo, ao usar [FatScript Playground](#)), o suporte limitado da plataforma para multithreading afeta a implementação do `Worker`. Para maximizar a compatibilidade do código entre plataformas, as tarefas do `Worker` são executadas inline e bloqueiam a thread principal quando o método `start` é chamado. Esta abordagem compromete as vantagens da execução assíncrona, mas permite uma implementação consistente entre plataformas em muitos dos cenários.

Veja também

- [Biblioteca Time](#)

bridge

Ponte entre FatScript e bibliotecas C externas

a biblioteca `bridge` permite que o FatScript interaja com bibliotecas C externas, fornecendo capacidades de linkagem dinâmica e interface de função estrangeira (FFI); isso é útil para aproveitar as vantagens de desempenho do C e usar bibliotecas C existentes em aplicativos FatScript

Importação

```
_ <- fat.bridge
```

Tipos

A biblioteca `bridge` introduz dois tipos principais para lidar com a linkagem dinâmica de bibliotecas externas e chamar funções externas: `DLL` e `FFI`.

DLL

O tipo `DLL` representa um identificador para uma biblioteca carregada dinamicamente.

Construtor

Nome	Assinatura	Breve descrição
<code>DLL</code>	<code>(filename)</code>	Carrega uma biblioteca dinâmica

O construtor `DLL` aceita o seguinte argumento:

- **filename:** O caminho para o arquivo de objeto compartilhado (`.so/.dll`) a ser carregado.

FFI

O tipo `FFI` permite vincular funções externas da biblioteca carregada dinamicamente, utilizando o sistema `CType` do FatScript para corresponder os tipos de entrada e saída esperados pela função.

Construtor

Nome	Assinatura	Resumo
<code>FFI</code>	<code>(lib, name, in, out)</code>	Vincula a uma função externa

O construtor `FFI` aceita os seguintes argumentos:

- **lib:** Uma instância de `DLL` representando a biblioteca carregada.
- **name:** Um nome (`Text`) da função a ser vinculada dentro da biblioteca.
- **in:** Uma lista `List/Ctype` de tipos de argumento esperados pela função.
- **out:** O tipo de retorno `CType` da função.

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>call</code>	<code>(args...)</code>	Any Chama a função vinculada

Aliases

- **CPointer:** Representa um ponteiro de memória (semelhante a `void*`), com o tipo base `Chunk`.
- **CType:** Representa tipos C no FatScript, com o tipo base `Number`.

CType

O sistema `CType` mapeia tipos C comuns para tipos correspondentes no FatScript, permitindo uma interação segura com bibliotecas C. Os seguintes tipos estão disponíveis no escopo `ctype` e implementam a correspondência automática com os

tipos FatScript:

Nome	Tipo C	Correspondência
sint	int	Number
sintP	int*	Number
uint	unsigned int	Number
uintP	unsigned int*	Number
float	float	Number
floatP	float*	Number
double	double	Number
doubleP	double*	Number
schar	char	Chunk
scharP	char*	Chunk
uchar	unsigned char	Chunk
ucharP	unsigned char*	Chunk
short	short	Number
shortP	short*	Number
ushort	unsigned short	Number
ushortP	unsigned short*	Number
slong	long	Number
slongP	long*	Number
ulong	unsigned long	Number
ulongP	unsigned long*	Number
string	char* (tipo retorno)	Text
void	void (tipo retorno)	Void
voidP	void*	Chunk

`string` deve ser um ponteiro de caractere alocado dinamicamente e terminado em nulo

Métodos avulsos

Nome	Assinatura	Breve descrição
<code>unsafePeek</code>	(ptr: CPointer, offset: Number, len: Number): Chunk	Lê de ptr, considerando deslocamento e comprimento
<code>detachNode</code>	(node: Any): Void	Libera à propriedade da memória
<code>marshal</code>	(val: Any, type: CType): Chunk	Converte um valor FatScript para um tipo C bruto
<code>unmarshal</code>	(raw: Chunk, type: CType): Any	Converte de um tipo C para FatScript
<code>getErrno</code>	<> Number	Retorna o errno da última chamada FFI
<code>sizeof</code>	(type: CType): Number	Obtém o número de bytes para um determinado CType

unsafePeek

Permite a leitura direta de memória bruta, que pode ser usada para interagir com estruturas de dados C. **Aviso:** Este método não realiza verificação de limites e depende de parâmetros corretos. O uso incorreto pode causar falhas no sistema ou vulnerabilidades de segurança.

detachNode

Renuncia à propriedade da memória apontada por `Text` ou `Chunk` para evitar a liberação dupla de memória. Consulte a documentação da biblioteca externa para entender a propriedade da memória antes de usar `detachNode`, já que nem sempre é necessário.

marshal

Converte um valor FatScript para um bloco de memória bruta usando um `CType` específico. Útil para compor structs C. Apenas `string` e `voidP` são válidos para marshaling de `Text` e `Chunk` como tipos de ponteiro, respectivamente. **Aviso:** Garanta o manuseio correto dos ponteiros de buffer para evitar liberação dupla de memória.

unmarshal

Converte blocos de memória bruta para tipos específicos do FatScript com base em CType. Útil para interpretar dados retornados de structs C. **Aviso:** O uso incorreto ou CType incorreto pode resultar em comportamento indefinido ou corrupção de dados.

getErrno

O `errno` da última chamada FFI é armazenado em cache e pode ser recuperado por este método.

sizeof

Determina o tamanho da memória (em bytes) de um determinado CType. Isso é útil para usar com segurança funções como `unsafePeek`.

Exemplo de Uso**Carregando uma biblioteca**

Para carregar uma biblioteca dinâmica, use o tipo DLL:

```
zlibDLL = DLL('libz.so')
```

Isso tentará carregar a biblioteca de objeto compartilhado `libz.so` (neste exemplo, a biblioteca de compressão `zlib`).

Vinculando a uma função

Para vincular a uma função dentro da biblioteca carregada, use o tipo FFI:

```
compressFFI = FFI(zlibDLL, 'compress', [ucharP, slongP, ucharP, slong], sint)
```

Isso vincula à função `compress` na biblioteca `zlib`. Os tipos de argumento e o tipo de retorno são especificados usando CType.

Chamando a função

Uma vez vinculada, você pode chamar a função usando o método `call`:

```
compressedData = compressFFI.call(destBuff, destSize, source, sourceSize)
```

Isso chama a função `compress` e retorna o resultado.

Exemplo Completo: comprimindo dados com zlib

```
_ <- fat.type._
bridge <- fat.bridge

zlibDLL = DLL('libz.so')

{ ucharP, slong, slongP, sint } = bridge.ctype

compressFFI = FFI(zlibDLL, 'compress', [ucharP, slongP, ucharP, slong], sint)

# Comprimir dados
source = 'Hello, zlib compression!'.toChunk
destSize = 256
destBuff = Chunk(256)

compressedData = compressFFI.call(destBuff, destSize, source, source.size)
```

Observe que `destSize` usa o mapeamento do tipo `slongP`, e embora seja considerado imutável no FatScript, ele pode ser alterado pela chamada da função. Esse comportamento é esperado e é a forma de interface do FatScript com o C.

Manipulação avançada de dados brutos

Para obter uma melhor compreensão de como o `bridge` funciona, você pode estudar o [caso de teste do FFI](#) e os projetos de implementação de exemplo [zlib.fat](#) e [gcode.fat](#).

bridge

Bridge na Web Build

Ao usar fry compilado com Emscripten (por exemplo, ao usar [FatScript Playground](#)), não há suporte para esta biblioteca.

color

Códigos de cores ANSI para console

Importação

```
_ <- fat.color
```

Constantes

- black, 0
- red, 1
- green, 2
- yellow, 3
- blue, 4
- magenta, 5
- cyan, 6
- white, 7
- bright.black, 8
- bright.red, 9
- bright.green, 10
- bright.yellow, 11
- bright.blue, 12
- bright.magenta, 13
- bright.cyan, 14
- bright.white, 15

Métodos

Nome	Assinatura	Breve descrição
detectDepth	<> Number	Obter suporte de cor do console
to8	(xr: Any, g: Number = \emptyset , b: Number = \emptyset)	Converter RGB para 8-cores
to16	(xr: Any, g: Number = \emptyset , b: Number = \emptyset)	Converter RGB para 16-cores
to256	(xr: Any, g: Number = \emptyset , b: Number = \emptyset)	Converter RGB para 256-cores
to24Bit	(xr: Any, g: Number = \emptyset , b: Number = \emptyset)	Codificar RGB para cores reais

Notas de uso

to8, to16, to256 e to24Bit

O parâmetro `xr` pode ser um texto opcional que representa a cor no formato HTML. Por exemplo, pode ser fornecido como `'fae830'` ou `'#fae830'` (amarelo):

```
color <- fat.console
console <- fat.console
```

```
console.log('hey', color.to16('fae830'))
console.log('hey', color.to256('fae830'))
```

No entanto, se `xr` for um número entre 0 e 255 representando `r`, então os parâmetros `g` e `b` serão necessários:

```
console.log('hey', color.to256(250, 232, 48)) // mesmo resultado
```

estes métodos podem produzir aproximações da cor original nas profundidades 8, 16 ou 256 e não a cor verdadeira exata

Veja também

- [Biblioteca console](#)
- [Biblioteca curses](#)
- [256 Cores](#)

console

Operações de entrada e saída do console

Importação

```
_ <- fat.console
```

Métodos

Nome	Assinatura	Breve descrição
log	(msg: Any, fg: Number = 0, bg: Number = 0): Void	Imprime msg para stdout, com quebra de linha
print	(msg: Any, fg: Number = 0, bg: Number = 0): Void	Imprime msg para stdout, sem quebra de linha
stderr	(msg: Any, fg: Number = 0, bg: Number = 0): Void	Imprime msg para stderr, com quebra de linha
input	(msg: Any, mode: Text = 0): Text	Imprime msg e retorna entrada de stdin
flush	<> Void	Esvazia buffer de saída padrão
cls	<> Void	Limpa stdout usando códigos de escape ANSI
moveTo	(x: Number, y: Number): Void	Move o cursor usando códigos de escape ANSI
isTTY	<> Boolean	Verifica se stdout é um terminal
showProgress	(label: Text, fraction: Number): Void	Renderiza barra de progresso, fração 0 a 1

os métodos `log`, `stderr` e `input` garantem segurança de threads em cenários assíncronos

Notas de uso

saída

Por padrão, `stdout` e `stderr` imprimem no console. Os parâmetros de cor de primeiro plano (`fg`) e cor de plano de fundo (`bg`) são opcionais.

as cores são automaticamente suprimidas se o buffer de saída não for um TTY

input

O parâmetro opcional `mode` aceita os seguintes valores:

- 'plain', entrada simples (sem cursor readline, sem histórico)
- 'quiet', como modo plain, porém sem feedback
- 'secret', modo especial para leitura de senha
- `null` (padrão), com readline e histórico de entrada

Veja também

- [Biblioteca color](#)
- [Biblioteca curses](#)

curses

Interface de usuário baseada em terminal

embora a inspiração seja reconhecida, o FatScript tem seu próprio modo de abordar a interface de usuário no terminal, que difere em muitos aspectos da biblioteca curses original

Importação

```
_ <- fat.curses
```

Métodos

Nome	Assinatura	Breve descrição
box	(p1: Scope, p2: Scope): Void	Desenhar quadrado de pos1 a pos2
fill	(p1: Scope, p2: Scope, p: Text = '): Void	Preencher de pos1 a pos2 com p
clear	<> Void	Limpar buffer de tela
refresh	<> Void	Renderizar buffer de tela
getMax	<> Scope	Retorna tamanho da tela como x, y
printAt	(pos: Scope, msg: Any, width: Number = 0): Void	Imprimir msg em { x, y } pos
makePair	(fg: Number = 0, bg: Number = 0): Number	Criar um par de cores
usePair	(pair: Number): Void	Aplicar par de cores
frameTo	(cols: Number, rows: Number)	Alinhar área ao centro da tela
setMouse	(enabled: Boolean): Void	Rastreamento de mouse com readKey
readKey	<> Text	Retorna tecla pressionada
readText	(pos: Scope, width: Number, prev: Text = 0): Text	Inicia caixa de texto
flushKeys	<> Void	Limpar buffer de entrada
endCurses	<> Void	Encerrar o modo curses

as posições (pos) estão no formato { x: Number, y: Number }

os métodos nesta biblioteca **não garantem** a segurança de threads em cenários assíncronos, ou utilize a thread principal **ou então** um único [worker](#) para renderizar atualizações no console

Notas de uso

Qualquer método desta biblioteca, exceto `getMax` e `endCurses`, iniciará o modo curses se ainda não tiver iniciado. Note que métodos como `log`, `stderr` e `input` da biblioteca [console](#) chamarão `endCurses` implicitamente. No entanto, `moveTo`, `print` e `flush` não irão alterar o modo de saída e podem ser combinados com métodos curses, o que pode ser útil em algumas circunstâncias.

As letras x e y representam coluna e linha, respectivamente, ao chamar `printAt`, onde (0, 0) é o canto superior esquerdo e o resultado de `getMax` é apenas a primeira coordenada fora do canto inferior direito.

caracteres especiais em curses só funcionam se um [locale](#) UTF-8 puder ser definido

makePair

Você pode importar a biblioteca [color](#) para usar nomes de cores e criar uma combinação de primeiro plano e plano de fundo (par). Passe `NULL` para aplicar a cor padrão no parâmetro desejado.

usePair

A entrada deste método deve ser um par de cores criado com o método `makePair`. Ele deixa este par habilitado até que você chame-o novamente com um par diferente.

readKey

Este método não bloqueia e retorna `null` se `stdin` estiver vazio, caso contrário retornará um caractere por vez.

Chaves especiais podem ser detectadas e retornar palavras-chave como:

- teclas de seta:
 - up
 - down
 - left
 - right
- teclas de edição:
 - delete
 - backspace
 - enter
 - space
 - tab
 - backTab (shift+tab)
- teclas de controle:
 - pageUp
 - pageDown
 - home
 - end
 - insert
 - esc
- outras:
 - resize (janela do terminal foi redimensionada)
 - mouse:botão:coluna:linha:éLiberação (quanto rastreando eventos do mouse)

a detecção correta das teclas pode depender do contexto ou da plataforma

readText

Entra em modo captura de texto utilizando uma área demarcada por posição e largura da caixa de texto. Se o texto for maior que o espaço uma rolagem automática do texto é realizada. O texto completo é retornado ao pressionar `enter` ou `tab`, no entanto caso `ESC` seja pressionado é retornado `null`.

Veja também

- [Tipo MouseEvent](#)
- [Biblioteca color](#)
- [Biblioteca console](#)

enigma

Métodos de criptografia, hash e UUID

Importação

```
_ <- fat.enigma
```

Métodos padrão

Estes métodos estão disponíveis em todas as construções `fry`. Embora `derive`, `encrypt` e `decrypt` ajudem a criar cifras "não legíveis por humanos", eles não são considerados seguros criptograficamente. **NÃO os use sozinhos para proteger dados sensíveis!**

Nome	Assinatura	Resumo
<code>getHash</code>	<code>(msg: Text): Number</code>	Calcula hash de 32 bits do texto
<code>genUUID</code>	<code><> Text</code>	Gera um UUID (versão 4)
<code>genKey</code>	<code>(len: Number): Text</code>	Gera uma chave aleatória
<code>derive</code>	<code>(secret: Text, salt: Text, iter: Number): Text</code>	Função básica de derivação de chave
<code>encrypt</code>	<code>(msg: Text, key: Text = ∅): Text</code>	Criptografa mensagem usando chave
<code>decrypt</code>	<code>(msg: Text, key: Text = ∅): Text</code>	Descryptografa mensagem usando chave

Métodos OpenSSL

Estes métodos são seguros criptograficamente e estão disponíveis apenas em builds que incluem suporte ao `OpenSSL`. Eles fornecem ferramentas robustas para lidar com criptografia, descryptografia, hash e derivação de chave com altos padrões de segurança.

Nome	Assinatura	Resumo
<code>digest</code>	<code>(data: Chunk, algo: Text = 'sha256'): HugeInt</code>	Calcula um hash seguro
<code>bytes</code>	<code>(len: Number): Chunk</code>	Gera bytes aleatórios
<code>pbkdf2</code>	<code>(secret: Text, salt: Text, iter: Number, algo: Text): HugeInt</code>	Deriva uma chave usando PBKDF2
<code>hmac</code>	<code>(data: Chunk, key: Chunk, algo: Text): HugeInt</code>	Calcula um código de autenticação
<code>encryptAES</code>	<code>(data: Chunk, key: Chunk): Chunk</code>	Criptografa com chave AES-256
<code>decryptAES</code>	<code>(data: Chunk, key: Chunk): Chunk</code>	Descryptografa dados em AES-256

Notas de uso

`getHash`

Um hash de 32 bits é suficiente para proteger contra corrupção de dados em até 100kb. Fry usa `FNV1A_Jesteress`, que é um dos algoritmos de hash mais rápidos e "suficientemente bons" para strings "longas".

`genUUID`

Um UUID, ou Identificador Único Universal, é um número de 128 bits usado para identificar objetos ou entidades em sistemas computacionais. A implementação gera UUIDs aleatórios seguindo o formato da especificação RFC 4122 versão 4, mas não adere estritamente aos padrões de aleatoriedade segura criptograficamente.

`genKey`

Gera uma chave aleatória usando o alfabeto Base64.

`derive (inseguro)`

Esta função determinística de derivação de chave produz uma string de 32 caracteres usando o alfabeto Base64. Embora possa parecer oferecer 192 bits de entropia, a entropia efetiva é **significativamente menor** devido à função de hash subjacente.

Nota: A intenção original desta função é também o alongamento de chave. Ela é projetada para ser usada em combinação com as funções `encrypt` e `decrypt`, oferecendo uma camada adicional de processamento para as chaves utilizadas nestas operações de criptografia. Porém, como o método não fornece segurança criptográfica robusta por si só, é recomendado o uso de métodos de derivação de chave mais seguros para aplicações que exijam altos padrões de proteção de dados.

encrypt/decrypt (inseguro)

Estas funções podem ser usadas com ou sem uma chave especificada (em branco para chave padrão). Elas empregam uma operação simples de XOR combinada com um hash de controle e são codificadas em Base64. Estes métodos **não são criptograficamente seguros** e não devem ser usados para proteger dados sensíveis, a menos que sejam combinados com uma [cifra de uso único](#).

Detalhes adicionais para métodos OpenSSL

digest

Suporta múltiplos algoritmos de hash, incluindo sha1, sha224, sha256, sha384, sha512, sha3_224, sha3_256, sha3_384, sha3_512, permitindo flexibilidade dependendo dos requisitos de segurança.

bytes

Ideal para criar chaves criptográficas de alta entropia e vetores de inicialização (IVs).

pbkdf2

Utiliza uma senha, sal e um número especificado de iterações junto com uma função de hash criptográfico para produzir uma chave de criptografia robusta.

hmac

Garante a integridade dos dados e autenticidade usando uma chave secreta e o algoritmo de hash especificado.

encryptAES/decryptAES

Operam usando o modo AES-256-CCM, proporcionando confidencialidade e autenticidade. A chave deve ter exatamente 32 bytes, obtida de uma fonte segura como a função `pbkdf2`, isso normalmente pode ser alcançado assim:

```
key: HugeInt = enigma.pbkdf2(...)
encryptionKey = key.toChunk.fit(32)
```


failure

Tratamento de erros e gerenciamento de exceções

Importação

```
_ <- fat.failure
```

Métodos

Nome	Assinatura	Breve descrição
trap	<> Void	Aplicar manipulador genérico de erros
trapWith	(handler: Method): Void	Definir um manipulador para erros no contexto
untrap	<> Void	Desarmar o manipulador para erros no contexto
noCrash	(unsafe: Method): Any	Continua em caso de erro dentro do método unsafe

Notas de uso

Quando um erro é gerado se um manipulador de erro for encontrado, buscando do contexto de execução interno para o externo, o manipulador que envolve a falha é invocado automaticamente com esse erro como argumento e o contexto de chamada é encerrado com o valor de retorno do manipulador de erro.

trapWith

Este método vincula um manipulador de erros ao contexto do site de chamada, por exemplo quando usado dentro de um método, ele protegerá a lógica executada dentro do corpo desse método, e se um erro ocorrer, o método será encerrado retornando o que quer que seja retornado pelo próprio manipulador de erros.

você pode precisar garantir que o seu manipulador de erros também retorne um tipo válido para esse contexto

Exemplo

Defina um manipulador de erro que imprima o erro e saia:

```
console <- fat.console
system  <- fat.system
sdk     <- fat.sdk

simpleErrorHandler = (error) -> {
  console.log(error)
  sdk.printStackTrace(10)
  system.exit(system.failureCode)
}
```

Finalmente, use o método `trapWith` para atribuir o manipulador de erro:

```
failure <- fat.failure
failure.trapWith(simpleErrorHandler)
```

Trap it!

Você pode lidar com erros esperados ou deixar passar o inesperado:

```
failure <- fat.failure
_       <- fat.type.Error

MyError = Error

errorHandler = (e): Number -> e >> {
  MyError => 0 # resolve (esperado)
  _      => e  # "pass through" (inesperado)
}
```

failure

```
unsafeMethod = (n): Number -> {  
  failure.trapWith(errorHandler)  
  n < 10 ? MyError('arg é menor que dez')  
  n - 10  
}
```

Neste caso, o programa não travará se você chamar `unsafeMethod(5)`, mas se você comentar a linha `trapWith`, verá que ele trava com `MyError`.

Veja também

- [Error \(sintaxe\)](#)
- [Extensões do protótipo Error](#)
- [Controle de fluxo](#)

file

Operações de entrada e saída de arquivo

Importação

```
_ <- fat.file
```

Contribuições de Tipo

Nome	Assinatura	Breve descrição
FileInfo	(modTime: Epoch, size: Text)	Metadados do arquivo

Métodos

Nome	Assinatura	Breve descrição
basePath	<> Text	Extrair caminho onde o app foi chamado
resolve	(path: Text): Text	Retorna nome canônico do caminho
exists	(path: Text): Boolean	Verificar se existe arquivo no caminho
read	(path: Text): Text	Ler arquivo do caminho (modo de texto)
readBin	(path: Text): Chunk	Ler arquivo do caminho (modo de binário)
readSys	(path: Text): Text	Ler arquivo de sistema/virtual do caminho
write	(path: Text, src): Void	Escrever src no arquivo e retornar sucesso
append	(path: Text, src): Void	Acrescentar ao arquivo e retornar sucesso
remove	(path: Text): Void	Apagar arquivos e diretórios
isDir	(path: Text): Boolean	Verificar se o caminho é um diretório
mkdir	(path: Text, safe: Boolean)	Criar um diretório
lsDir	(path: Text): List/Text	Obter lista de arquivos em um diretório
stat	(path: Text): FileInfo	Obter metadados do arquivo

a partir da versão 3.3.0, em caso de exceção, todos os métodos da biblioteca `file` lançam `FileError` no lugar de retornar um valor booleano ou nulo, oferecendo uma interface mais homogênea com as demais bibliotecas padrão

Notas de uso

write/append

Estes métodos tratam de forma inteligente diferentes tipos de dados para otimizar a saída de arquivos. Para o tipo `Chunk`, eles escrevem automaticamente em modo binário e para o tipo `Text`, como texto simples. Para outros tipos, eles implicitamente "stringificam" o valor `src` antes de escrever, garantindo que todos e qualquer valor seja tratado de maneira adequada.

remove

O comportamento é semelhante a `rm -r`, removendo arquivos e diretórios de forma recursiva.

a partir da versão 3.0.1, links simbólicos não são seguidos; na versão 3.0.0, os links simbólicos eram seguidos; as versões anteriores não implementavam exclusão recursiva

mkdir

O comportamento é semelhante a `mkdir -p` criando diretórios intermediários quando necessário.

Se `safe` for definido como `true`, o novo diretório receberá permissões 0700, oferecendo mais proteção, em vez das permissões padrão 0755, que oferecem menos proteção.

read vs. readSys

file

O método `read` é otimizado para a leitura de arquivos regulares com tamanhos previsíveis, utilizando `stat` para alocar memória de forma eficiente antes de ler o arquivo inteiro. Em contraste, `readSys` é projetado para arquivos de sistema ou virtuais de diretórios como `/proc` ou `/sys`, onde os tamanhos dos arquivos não podem ser determinados de antemão. Ele ajusta a alocação de memória dinamicamente durante a leitura.

Veja também

- [Biblioteca failure](#)
- [Biblioteca recode](#)

http

Framework de manipulação HTTP

Importação

```
_ <- fat.http
```

Route

Uma rota é uma estrutura usada para mapear métodos HTTP para certos padrões de caminho, especificando qual código deve ser executado quando uma requisição é recebida. Cada rota pode definir um comportamento diferente para cada método HTTP (POST, GET, PUT, DELETE).

Construtor

Nome	Assinatura	Breve descrição
Route	(path: Text, post: Method, get: Method, put: Method, delete: Method)	Constrói um objeto Route

cada método implementado recebe um `HttpRequest` como argumento e deve retornar um objeto `HttpResponse`

HttpRequest

Um `HttpRequest` representa uma mensagem de requisição HTTP. Isso é o que seu servidor recebe de um cliente quando ele faz uma requisição ao seu servidor.

Construtor

Nome	Assinatura	Breve descrição
HttpRequest	(method: Text, path: Text, headers: List, params: Scope, body: Any)	Constrói um objeto HttpRequest

os items `params` e `body` podem ser omitidos dependendo da requisição recebida

HttpResponse

Um `HttpResponse` representa uma mensagem de resposta HTTP. Isso é o que um servidor envia de volta ao cliente em resposta a uma requisição HTTP.

para respostas em modo cliente, `body` será provido como `Text` se um tipo MIME textual for encontrado nos `headers`, caso contrário, será fornecido como `Chunk`

Construtor

Nome	Assinatura	Breve descrição
HttpResponse	(status: Number, headers: List/Text, body: Any)	Constrói um objeto HttpResponse

Métodos

Nome	Assinatura	Breve descrição
setHeaders	(headers: List): Void	Definir cabeçalhos de solicitações
post	(url: Text, body, wait): HttpResponse	Criar/postar body para url
get	(url: Text, wait): HttpResponse	Ler/obter de url
put	(url: Text, body, wait): HttpResponse	Atualizar/colocar body na url
delete	(url: Text, wait): HttpResponse	Excluir em url
setName	(name: Text): Void	Definir agente/nome do servidor
verifySSL	(enabled: Boolean): Void	Configuração SSL (modo cliente)
setSSL	(certPath: Text, keyPath: Text): Void	Configuração SSL (modo servidor)
listen	(port: Number, routes: List/Route, maxMs)	Provedor de endpoint (modo servidor)

http

`body`: Any e `wait`: Number são sempre parâmetros opcionais, sendo que caso `body` não se enquadre como `Text` ou `Chunk` será automaticamente convertido para JSON no processo de envio e `wait` é o tempo máximo de espera e o padrão é 30.000ms (30 segundos)

`verifySSL` está habilitado por padrão para o modo cliente

`setSSL` pode não estar disponível, caso o sistema não tenha a biblioteca OpenSSL

Notas de uso

Modo cliente

Em `HttpResponse`. `body`, você pode precisar decodificar explicitamente uma resposta JSON para `Scope` usando o método `fromJSON`. Para postar um tipo nativo como JSON, você pode codificá-lo usando o método `toJSON`; no entanto, isso não é estritamente necessário, pois será feito implicitamente. Ambos os métodos estão disponíveis na biblioteca [fat.recode](#).

Se os cabeçalhos não forem definidos, o cabeçalho `Content-Type` padrão para `Chunk` será `application/octet-stream`, para `Text` será `text/plain`; `charset=UTF-8` e para outros tipos, será `application/json`; `charset=UTF-8` (devido à conversão implícita).

Você pode definir cabeçalhos de solicitação personalizados da seguinte forma:

```
http <- fat.http

url = ...
token = ...
body = ...

http.setHeaders([
  "Accept: application/json; charset=UTF-8"
  "Content-Type: application/json; charset=UTF-8"
  "Authorization: Bearer " + token # cabeçalhos personalizado
])

http.post(url, body)
```

definir cabeçalhos substituirá completamente a lista anterior pela nova lista

Ao realizar solicitações assíncronas, você pode precisar chamar `setHeaders`, `setName` e configurar `verifySSL` dentro de cada `Worker`, já que essas configurações são locais para cada `thread`.

Modo servidor

Lidando com respostas HTTP

Você pode definir o parâmetro opcional `maxMs` ao chamar `listen` para restringir quanto tempo o servidor vai esperar para que cada solicitação transfira seu conteúdo (conexão de entrada), retornando status 408 caso excedido.

O servidor `FatScript` lida automaticamente com os códigos de status HTTP comuns, como 200, 400, 404, 405, 408, 500 e 501. Sendo 200 o padrão ao construir um objeto `HttpResponse`.

Além dos códigos de status retornados de forma automática, você também pode retornar explicitamente outros códigos de status, como 201, 202, 203, 204, 205, 301, 401 e 403, especificando o código de status no objeto `HttpResponse`, por exemplo: `HttpResponse(status = 401)`. Para todos os códigos mencionados aqui, o servidor fornece corpos de resposta padrão em texto simples. No entanto, você tem a opção de substituir esses valores padrão e fornecer seus próprios corpos de resposta personalizados, quando necessário.

Ao lidar automaticamente com esses códigos de status e fornecer corpos de resposta padrão, o servidor `FatScript` simplifica o processo de desenvolvimento, ao mesmo tempo em que permite que você tenha controle sobre o conteúdo da resposta quando necessário.

não pertencendo a nenhum dos códigos anteriores, o servidor irá retornar o código 500

Veja um exemplo de um simples servidor HTTP de arquivos:

```
_ <- fat.std
# adapte para o local do conteúdo
```

http

```
directory = '/home/user/contentFolder'

# restrito a algumas extensões somente
mediaTypesByExtension = {
  htm = html = 'text/html'
  js = 'application/javascript'
  json = 'application/json'
  css = 'text/css'
  md = 'text/markdown'
  xml = 'application/xml'
  csv = 'text/csv'
  txt = 'text/plain'
  svg = 'image/svg+xml'
  rss = 'application/rss+xml'
  atom = 'application/atom+xml'
  png = 'image/png'
  jpg = jpeg = 'image/jpeg'
  gif = 'image/gif'
  ico = 'image/x-icon'
  webp = 'image/webp'
  woff = 'font/woff'
  woff2 = 'font/woff2'
}

routes: List/Route = [
  Route(
    '*'
    get = (request: HttpRequest): HttpResponse -> {
      path = file.resolve(directory + request.path) # sanitized path
      type = path & path.startsWith(directory) # jailed access
      ? mediaTypesByExtension(path.split('.').(-1))

      path.isEmpty => HttpResponse(status = 404) # not found
      type.isEmpty => HttpResponse(status = 403) # forbidden
      => HttpResponse(
        - status = 200
          headers = [ 'Content-Type: {type}' ]
          body = file.readBin(path)
      )
    }
  )
]

http.listen(8080, routes)
```

use `http.listen(0, routes)` para iniciar um servidor com um número de porta atribuído automaticamente e verifique a porta efetiva atribuída ao servidor através do comando embutido `$port`

math

Operações e funções matemáticas

Importação

```
_ <- fat.math
```

Constantes

- e, logaritmo natural constante 2.71...
- maxInt, 9007199254740992
- minInt, -9007199254740992
- pi, razão do círculo para o seu diâmetro 3.14...

leia mais sobre [precisão numérica](#) no FatScript

Funções básicas

Nome	Assinatura	Breve descrição
abs	(x: Number): Number	Retorna o valor absoluto de x
ceil	(x: Number): Number	Retorna o menor inteiro $\geq x$
floor	(x: Number): Number	Retorna o maior inteiro $\leq x$
isInf	(x: Number): Boolean	Retorna true se x for infinito
isNaN	(x: Any): Boolean	Retorna true se x não for um número
logN	(x: Number, base: Number = e): Number	Retorna o logaritmo
random	$\langle \rangle$ Number	Retorna pseudo-aleatório, onde $0 \leq n < 1$
sqrt	(x: Number): Number	Retorna a raiz quadrada de x
round	(x: Number): Number	Retorna o inteiro mais próximo de x

Funções trigonométricas

Nome	Assinatura	Breve descrição
sin	(x: Number): Number	Retorna o seno de x
cos	(x: Number): Number	Retorna o cosseno de x
tan	(x: Number): Number	Retorna a tangente de x
asin	(x: Number): Number	Retorna o arco seno de x
acos	(x: Number): Number	Retorna o arco cosseno de x
atan	(x: Number, y = 1): Number	Retorna o arco tangente de x, y
radToDeg	(r: Number): Number	Converte radianos para graus
degToRad	(d: Number): Number	Converte graus para radianos

Funções hiperbólicas

Nome	Assinatura	Breve descrição
sinh	(x: Number): Number	Retorna o seno hiperbólico de x
cosh	(x: Number): Number	Retorna o cosseno hiperbólico de x
tanh	(x: Number): Number	Retorna a tangente hiperbólica de x

Funções estatísticas

Nome	Assinatura	Breve descrição
mean	(v: List/Number): Number	Retorna a média de um vetor
median	(v: List/Number): Number	Retorna a mediana de um vetor

Nome	Assinatura	Breve descrição
sigma	(v: List/Number): Number	Retorna o desvio padrão de um vetor
variance	(v: List/Number): Number	Retorna a variância de um vetor
max	(v: List/Number): Number	Retorna o valor máximo no vetor
min	(v: List/Number): Number	Retorna o valor mínimo no vetor
sum	(v: List/Number): Number	Retorna a soma do vetor

Outras funções

Nome	Assinatura	Breve descrição
fact	(x: Number): Number	Retorna o fatorial de x
exp	(x: Number): Number	Retorna e elevado à potência de x
sigmoid	(x: Number): Number	Retorna o sigmoid de x
relu	(x: Number): Number	Retorna o ReLU de x

Exemplo

```
math <- fat.math # importação nomeada
math.abs(-52)   # retorna 52
```

Veja também

- [Number \(sintaxe\)](#)
- [Extensões do protótipo Number](#)

recode

Conversão de dados entre vários formatos

Importação

```
_ <- fat.recode
```

o [pacote type](#) é automaticamente importado com esta importação

Variáveis

Estas configurações podem ser usadas para ajustar o comportamento das funções de processamento:

- csvSeparator, o padrão é , (vírgula)
- csvQuote, o padrão é " (aspas duplas)

Funções Base64

Nome	Assinatura	Breve descrição
toBase64	(data: Chunk): Text	Codifica bloco binário para texto base64
fromBase64	(b64: Text): Chunk	Decodifica texto base64 para formato original

Funções JSON

Nome	Assinatura	Breve descrição
toJSON	(val: Any): Text	Codifica JSON a partir de tipos nativos
fromJSON	(json: Text): Any	Decodifica JSON para tipos nativos

com toJSON os tipos nativos como HugeInt, Method e Chunk serão convertidos em null, enquanto Errors serão convertidos para texto

Funções URL

Nome	Assinatura	Breve descrição
toURL	(text: Text): Text	Codifica texto para texto escapado URL
fromURL	(url: Text): Text	Decodifica texto escapado URL para formato original
toFormData	(data: Scope): Text	Codifica escopo para Dados de Formulário URL
fromFormData	(data: Text): Scope	Decodifica Dados de Formulário URL para escopo

Funções CSV

Nome	Assinatura	Breve descrição
toCSV	(header: List/Text, rows: List/Scope): Text	Codifica CSV a partir de linhas
fromCSV	(csv: Text): List/Scope	Decodifica CSV para linhas

a partir da versão 4. x . x os métodos CSV oferecem suporte a aspas automáticas, aspas de escape, separadores e novas linhas dentro de aspas

Funções RLE

Nome	Assinatura	Breve descrição
toRLE	(chunk: Chunk): Chunk	Comprime para esquema RLE
fromRLE	(chunk: Chunk): Chunk	Descomprime de esquema RLE

Cópias congeladas e aquecidos

Nome	Assinatura	Breve descrição
<code>toFrostCopy</code>	<code>(item: Any): Any</code>	Cria uma cópia imutável do item
<code>toHotCopy</code>	<code>(item: Any): Any</code>	Cria uma cópia mutável do item

`toFrostCopy` garante imutabilidade, sendo ideal para capturar instantâneos de dados aninhados "seguros", enquanto `toHotCopy` permite que os dados sejam aquecidos novamente, útil em cenários onde dados congelados precisam de processamento adicional

Outras funções

Nome	Assinatura	Breve descrição
<code>inferType</code>	<code>(val: Text): Any</code>	Converte texto em void/boolean/number
<code>minify</code>	<code>(src: Text): Text</code>	Minifica código fonte FatScript

`minify` substituirá quaisquer instruções `$break` (ponto de interrupção do depurador) por `()`

Para desativar a inferência de tipo fornecida por `inferType` para `fromFormData` e `fromCSV`, você pode sobrescrevê-la globalmente usando `recode.inferType = -> _` após importar `fat.recode`, ou para reativar `recode.inferType = val -> $inferType`.

Veja também

- [Pacote type](#)
- [Biblioteca SDK](#)

sdk

Utilitários do kit de desenvolvimento de software do fry

uma biblioteca especial que expõe alguns dos elementos internos do interpretador fry

Importação

```
_ <- fat.sdk
```

Métodos

Nome	Assinatura	Breve descrição
ast	(<code>_</code>): Void	Imprime árvore de sintaxe abstrata do nó
stringify	(<code>_</code>): Text	Serializa o nó em texto semelhante a JSON
eval	(<code>_</code>): Any	Interpreta texto como programa FatScript
getVersion	<> Text	Retorna versão do fry
printStack	(depth: Number): Void	Imprime pilha do contexto de execução
readLib	(ref: Text): Text	Retorna o código-fonte da biblioteca fry
typeof	(<code>_</code>): Text	Retorna o nome do tipo do nó
getTypes	<> List	Retorna info sobre tipos declarados
getDef	(name: Text): Any	Retorna definição de tipo por nome
getMeta	<> Scope	Retorna os metadados do interpretador
setKey	(key: Text): Void	Definir chave para pacotes ofuscados
setMem	(n: Number): Void	Definir limite de memória (contagem de nós)
runGC	<> Number	Rodar o GC, retorna transcorrido em milissegundos
quickGC	<> Number	Roda uma única iteração do GC e retorna ms

Notas de uso

stringify

Enquanto `recode.toJSON` produz um JSON estritamente válido, `stringify` é mais flexível. Ele é capaz de exportar `HugeInt` como números hexadecimais (por exemplo, `0x123abc`), `Chunk` como codificado em Base64, e outros tipos também podem ter representações mais informativas do que apenas `null`. Essas representações são projetadas para permitir uma exportação mais rica para o ambiente FatScript e não são destinadas à serialização compatível com JSON.

readLib

```
_ <- fat.sdk
_ <- fat.console
```

```
print(readLib('fat.extra.Date')) # imprime a implementação da biblioteca Date
```

`readLib` não pode ver arquivos externos, mas `read` da [biblioteca file](#) pode

setKey

Use preferencialmente no arquivo `.fryrc` assim:

```
_ <- fat.sdk
setKey('secret') # irá codificar e decodificar pacotes com esta chave
```

Veja mais sobre [ofuscação](#).

setMem

Use preferencialmente no arquivo `.fryrc` assim:

sdk

```
_ <- fat.sdk  
setMem(5000) # ~2mb
```

Escolhendo entre GC completo e rápido

A maioria dos scripts simples em FatScript não precisará se preocupar com a gestão de memória, pois as configurações padrão são projetadas para fornecer uma capacidade de memória ampla e um comportamento automático eficiente desde o início. Geralmente, a melhor maneira de otimizar a performance é simplesmente ajustando o limite de memória. Em alguns casos raros, como em um loop de jogo ou processos iterativos complexos, você pode se beneficiar de chamar explicitamente o GC.

O método `quickGC` realiza uma limpeza rápida e menos exaustiva, tornando-o adequado para cenários onde é aceitável uma certa flexibilidade na alocação de memória. Por outro lado, `runGC` garante uma coleta de lixo completa e determinística, mas pode resultar em tempos de execução mais longos, dependendo de fatores como o tamanho e a complexidade do grafo de memória. No entanto, `quickGC` pode levar ao acúmulo de memória não reclamada, tornando-o menos eficaz em certos contextos. A melhor maneira de determinar a opção mais adequada é realizar testes comparativos em sua aplicação, simulando cenários de uso real.

rode seu script com a flag `-C` para avaliar o desempenho de sua execução

Veja mais sobre [gerenciamento de memória](#).

Veja também

- [Biblioteca recode](#)

smtp

Framework de manipulação SMTP

esta biblioteca fornece uma interface simples para configurar parâmetros SMTP e enviar e-mails

Importação

```
_ <- fat.smtp
```

Tipos

A biblioteca `smtp` introduz o tipo `ContactInfo`.

ContactInfo

O tipo `ContactInfo` representa um contato de e-mail, que pode incluir um nome opcional junto com o endereço de e-mail.

Construtor

Nome	Assinatura	Breve descrição
<code>ContactInfo</code>	<code>(email: Text, name: Text = "")</code>	Constrói um objeto <code>ContactInfo</code>

- **email**: O endereço de e-mail do contato.
- **name** (opcional): O nome do contato.

Métodos

config

Configura as definições de SMTP.

Parâmetro	Tipo	Descrição
<code>from</code>	<code>ContactInfo</code>	Um objeto que representa o remetente.
<code>server</code>	<code>Text</code>	A URL/endereço do servidor SMTP.
<code>username</code>	<code>Text</code>	O nome de usuário para autenticação SMTP.
<code>password</code>	<code>Text</code>	A senha para autenticação SMTP.
<code>useSSL</code>	<code>Boolean</code>	Usar SSL/TLS (padrão é <code>true</code>).

gera um erro se a configuração falhar

send

Envia um e-mail.

Parâmetro	Tipo	Descrição
<code>to</code>	<code>List/ContactInfo</code>	Uma lista de destinatários.
<code>subject</code>	<code>Text</code>	O assunto do e-mail.
<code>body</code>	<code>Text</code>	O corpo do e-mail.

retorna o UUID da mensagem em caso de sucesso

Notas de uso

Exemplo:

```
smtp <- fat.smtp
```

```
smtp.config(
  from = ContactInfo('sender@example.com', 'Nome do Remetente')
```

smtp

```
server = 'smtps://smtp.exemplo.com:porta'
username = 'seu_usuario'
password = 'sua_senha'
)

smtp.send(
  to = [
    ContactInfo('recipient1@example.com', 'Destinatário Um')
    ContactInfo('recipient2@example.com') # nome é opcional
  ]
  subject = 'E-mail de Teste'
  body = 'Este é um e-mail de teste enviado usando fat.smtp.'
)
```

SSL/TLS é ativado por padrão na configuração SMTP. Se o seu servidor SMTP exigir SSL/TLS, nenhuma configuração adicional é necessária. No entanto, se o seu servidor não suportar SSL/TLS, você pode desativá-lo definindo `useSSL` como `false` ao chamar `config`.

SMTP na Web Build

Ao usar `fry` compilado com Emscripten (por exemplo, ao usar [FatScript Playground](#)), não há suporte para esta biblioteca.

system

Operações e informações em nível de sistema

Importação

```
_ <- fat.system
```

Alias

ArgValues: uma lista de argumentos (Text) da linha de comando

Tipos

Nome	Assinatura	Breve descrição
CommandResult	(code: ExitCode, out: Text)	Tipo de retorno de capture

Constantes

- successCode, 0: ExitCode
- failureCode, 1: ExitCode

Métodos

Nome	Assinatura	Breve descrição
args	<> ArgValues	Retorna lista de argumentos passados pelo shell
exit	(code: Number): *	Sair do programa com o código de saída fornecido
getEnv	(var: Text): Text	Obter o valor da variável env por nome
shell	(cmd: Text): ExitCode	Executa cmd no shell, retorna o código de saída
capture	(cmd: Text): CommandResult	Captura a saída da execução de cmd
fork	(args: List/Text, out: Text = ∅)	Inicia um processo em segundo plano, retorna PID
kill	(pid: Number): Void	Envia SIGTERM para o processo pelo PID
getLocale	<> Text	Obter configuração de localidade atual
setLocale	(_: Text): Number	Definir configuração de localidade atual
getMacId	<> Text	Obter identificador da máquina (endereço MAC)
blockSig	(enabled: Boolean): Void	Bloqueia SIGINT, SIGHUP e SIGTERM

Notas de uso

Atenção!

É importante agir com cautela e responsabilidade ao utilizar os métodos `getEnv`, `shell`, `capture`, `fork` e `kill`. A biblioteca `system` oferece a capacidade de executar comandos diretamente do sistema operacional, o que pode introduzir riscos de segurança se não forem utilizadas com cuidado.

Para mitigar vulnerabilidades, evite utilizar a entrada do usuário diretamente na construção de comandos passados para esses métodos. A entrada do usuário deve ser validada para prevenir ataques de injeção de comandos e outras violações de segurança.

Tratamento de sinais

Quando ocorre uma interrupção Ctrl+C, a thread principal do interpretador FatScript captura o sinal e inicia um processo de limpeza. Durante esse processo, se ele detectar quaisquer [Workers](#) em execução, ele os terminará à força para evitar que o aplicativo fique travado.

Para aplicações que exigem um controle mais refinado sobre o processo de encerramento, o FatScript oferece uma opção para bloquear o tratamento padrão de sinais configurando `system.blockSig(true)`. Quando ativado, o interpretador não

system

capturará o Ctrl+C. Isso exige que você implemente seus próprios mecanismos de término, possivelmente através de `curses.readKey` ou outro método.

Outras Limitações (multithreading)

Embora os métodos desta biblioteca suportem uma variedade de tarefas de programação, eles não são otimizados para uso intercalado dentro de [Workers](#) assíncronos. Ao iniciar processos de dentro de threads, opte pelos métodos `shell/capture`, ou use exclusivamente `fork/kill`. Misturar esses dois pares de métodos em aplicações multithread pode resultar em comportamentos imprevisíveis.

em cada chamada, `shell/capture` definirá `SIGCHLD` para seu comportamento padrão, enquanto `fork` ignorará esse sinal para tentar evitar processos zumbis

fork

O parâmetro `out` permite redirecionar a saída padrão (`stdout`) para um arquivo de saída especificado. Se desejar descartar essa saída, você pode usar `"/dev/null"` como argumento.

get/set locale

O interpretador `fry` tentará inicializar o locale `LC_ALL` para `C.UTF-8` e, se esse locale não estiver disponível no sistema, tentará usar `en_US.UTF-8`, caso contrário, usará o locale padrão.

Veja mais sobre [nomes de localidade](#).

a configuração de localidade aplica-se apenas ao aplicativo e não é mantida após a saída do `fry`

time

Manipulação de hora e data

Importação

```
_ <- fat.time
```

[tipo Number](#) é importado automaticamente com esta importação

Métodos

Nome	Assinatura	Breve descrição
setZone	(offset: Number): Void	Definir fuso em milissegundos
getZone	<> Number	Obter diferença de fuso atual
now	<> Epoch	Obtenha o UTC atual em Epoch
format	(date: Text, fmt: Text = \emptyset): Epoch	Converter data para Epoch
parse	(date: Text, fmt: Text = \emptyset): Epoch	Converter Epoch em formato de data
wait	(ms: Number): Void	Aguardar milissegundos (suspender)
getElapsed	(since: Epoch): Text	Retorna o tempo decorrido como texto

Notas de uso

Epoch

No FatScript, o tempo é representado como um tipo aritmético para que você possa fazer contas.

Você pode obter o tempo decorrido entre tempo1 e tempo2 como:

```
decorrido = tempo2 - tempo1
```

Você também pode verificar se tempo2 acontece após tempo1, simplesmente assim:

```
tempo2 > tempo1
```

format

Formata a data em texto como "%Y-%m-%d %H:%M:%S.milliseconds" (padrão), quando `fmt` é omitido.

milissegundos só podem ser transformados no formato padrão, caso contrário, a precisão é de até segundos

parâmetro fmt

A especificação de formato é um texto contendo uma sequência de caracteres especiais chamada especificações de conversão, cada uma das quais é introduzida por um caractere '%' e terminada por algum outro caractere conhecido como especificador de conversão. Todos os outros caracteres são tratados como texto comum.

Especificador	Significado
%a	Nome abreviado do dia da semana
%A	Nome completo do dia da semana
%b	Nome do mês abreviado
%B	Nome completo do mês
%c	Data/Hora no formato da localidade
%C	Número do século [00-99], o ano dividido por 100 e truncado para um número inteiro
%d	Dia do mês [01-31]
%D	Formato de data, igual a %m/%d/%y
%e	O mesmo que %d, exceto que um único dígito é precedido por um espaço [1-31]
%g	Parte do ano de 2 dígitos da data da semana ISO [00,99]

Especificador	Significado
%F	Formato de data ISO, igual a %Y-%m-%d
%G	Parte do ano de 4 dígitos da data da semana ISO
%h	O mesmo que %b
%H	Hora no formato de 24 horas [00-23]
%I	Hora em formato de 12 horas [01-12]
%j	Dia do ano [001-366]
%m	Mês [01-12]
%M	Minuto [00-59]
%n	Caractere de nova linha
%p	Cadeia AM ou PM
%r	Hora no formato AM-PM da localidade
%R	Formato de 24 horas sem segundos, igual a %H:%M
%S	Segundo [00-61], o intervalo de segundos permite um segundo bissexto e um segundo bissexto duplo
%t	Caractere de tabulação
%T	Formato de 24 horas com segundos, igual a %H:%M:%S
%u	Dia da semana [1,7], segunda-feira é 1 e domingo é 7
%U	Número da semana do ano [00-53], domingo é o primeiro dia da semana
%V	Número da semana ISO do ano [01-53]. Segunda-feira é o primeiro dia da semana. Se a semana contendo 1º de janeiro tiver quatro ou mais dias no ano novo, será considerada a semana 1. Caso contrário, será a última semana do ano anterior e o ano seguinte será a semana 1 do ano novo.
%w	Dia da semana [0,6], domingo é 0
%W	Número da semana do ano [00-53], segunda-feira é o primeiro dia da semana
%x	Data no formato da localidade
%X	Hora no formato da localidade
%y	Ano de 2 dígitos [00,99]
%Y	Ano de 4 dígitos (pode ser negativo)
%z	String de deslocamento UTC com formato +HHMM ou -HHMM
%Z	Nome do fuso horário
%%	Caractere %

Sob o capô `format` usa C's [strftime](#) e `parse` usa C's [strptime](#), mas a tabela de especificação de formato acima se aplica praticamente nos dois sentidos.

type._

type._

Extensões do protótipo para [tipos nativos](#):

- [Void](#)
- [Boolean](#)
- [Number](#)
- [HugeInt](#)
- [Text](#)
- [Method](#)
- [List](#)
- [Scope](#)
- [Error](#)
- [Chunk](#)

FatScript **não** carrega essas definições automaticamente no escopo global, portanto você deve **explicitamente** [importar](#) quando necessário

Importando

Se você quiser disponibilizar todos eles de uma só vez, basta escrever:

```
_ <- fat.type._
```

...ou importe um por um, conforme necessário, por exemplo:

```
_ <- fat.type.List
```

características comum

Todos os tipos neste pacote suportam os seguintes métodos de protótipo:

- `apply` (construtor)
- `isEmpty`
- `nonEmpty`
- `size`
- `toText`

Exceto `Void` todos demais tipos implementam também o método `freeze`.

Veja também

- [Tipos \(sintaxe\)](#)

Void

Extensões do protótipo Void

Importação

```
_ <- fat.type.Void
```

Construtor

Nome	Assinatura	Breve descrição
Void	(val: Any)	Retorna nulo, apenas ignora o argumento

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro, sempre
nonEmpty	<> Boolean	Retorno falso, sempre
size	<> Number	Retorna 0, sempre
toText	<> Text	Retorna 'null' como texto

Exemplo

```
_ <- fat.type.Void  
x.isEmpty # true, já que x não foi declarado
```

Veja também

- [Void \(sintaxe\)](#)
- [Pacote type](#)

Boolean

Extensões do protótipo Boolean

Importação

```
_ <- fat.type.Boolean
```

Construtor

Nome	Assinatura	Breve descrição
Boolean	(val: Any)	Coage o valor para booleano

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro se falso
nonEmpty	<> Boolean	Retorna falso se verdadeiro
size	<> Number	Retorna 1 se verdadeiro, 0 se falso
toText	<> Text	Retorna 'true' ou 'false' como texto
freeze	<> Void	Torna o valor imutável

Exemplos

```
_ <- fat.type.Boolean

~ x = true
x.isEmpty      # false, já que x é verdadeiro
x.nonEmpty     # true, já que x não é vazio
x.size         # 1, já que verdadeiro é mapeado para tamanho 1
x.toText       # 'true', converte o booleano para texto

x.freeze
x = false      # gera um erro, já que x se tornou imutável após o freeze

Boolean('false') # retorna true, porque o texto é não-vazio
Boolean('')      # retorna false, porque é vazio
```

note que o construtor não tenta converter o valor do texto, o que é consistente com as avaliações de controle de fluxo, e você pode usar um simples [case](#) se precisar fazer conversão de texto para booleano

Veja também

- [Boolean \(sintaxe\)](#)
- [Pacote type](#)

Number

Extensões do protótipo Number

Importação

```
_ <- fat.type.Number
```

Aliases

- Epoch: tempo de época do unix em milissegundos
- ExitCode: status de saída ou código de retorno
- Millis: duração em milissegundos

Construtor

Nome	Assinatura	Breve descrição
Number	(val: Any)	Texto para número ou tamanho da coleção

realiza a conversão de texto para número assumindo a base decimal

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro se zero
nonEmpty	<> Boolean	Retorna verdadeiro se não-zero
size	<> Number	Retorna valor absoluto, igual a math.abs
toText	<> Text	Retorna número como texto
freeze	<> Void	Torna o valor imutável
format	(fmt: Text): Text	Retorna número como texto formatado
truncate	<> Number	Retorna número descartando decimais

Exemplo

```
_ <- fat.type.Number
x = Number('52') # número: 52
x.toText         # texto: '52'
x.format('.2')   # texto: '52.00'
```

format

O método `format` é usado para converter números em strings de várias maneiras. A estrutura básica de um especificador de formato é `%[flags][width][.precision][type]`. Aqui está o que cada um destes componentes significa:

- `flags` são caracteres opcionais que controlam o comportamento específico de formatação. Por exemplo, `0` pode ser usado para preenchimento com zeros e `-` para justificação à esquerda.
- `width` é um número inteiro que especifica o número mínimo de caracteres a serem impressos. Se o valor a ser impresso for mais curto do que este número, o resultado é preenchido com espaços em branco ou zeros, dependendo da flag utilizada.
- `precision` é um número opcional que segue um `.` que especifica o número de dígitos a serem impressos após o ponto decimal.
- `type` é um caractere que especifica como o número deve ser representado. Os tipos comuns são `f` (notação de ponto fixo), `e` (notação exponencial), `g` (fixo ou exponencial dependendo da magnitude do número) e `a` (notação de ponto flutuante hexadecimal).

Exemplos:

Number

- `%5.f`: Isso imprimirá o número com uma largura total de 5 caracteres, sem dígitos após o ponto decimal (porque a precisão é `f`, que significa ponto fixo, mas nenhum número segue o ponto). Será justificado à direita porque nenhuma flag `-` é usada.
- `%05.f`: Semelhante ao anterior, mas como a flag `0` é usada, os espaços vazios serão preenchidos com zeros.
- `%8.2f`: Isso imprimirá o número com uma largura total de 8 caracteres, com 2 dígitos após o ponto decimal.
- `%-8.2f`: Semelhante ao anterior, mas o número será justificado à esquerda por causa da flag `-`.
- `%.2e`: Isso imprimirá o número usando notação exponencial, com 2 dígitos após o ponto decimal.
- `%.2a`: Isso imprimirá o número usando notação de ponto flutuante hexadecimal, com 2 dígitos após o ponto hexadecimal.
- `%.2g`: Isso imprimirá o número em notação de ponto fixo ou exponencial, dependendo da sua magnitude, com no máximo 2 dígitos significativos.

caso o `%` não esteja presente, `fmt` é automaticamente avaliado como `%<fmt>f`

Veja também

- [Number \(sintaxe\)](#)
- [Biblioteca math](#)
- [Pacote type](#)

HugeInt

Extensões do protótipo HugeInt

Importação

```
_ <- fat.type.HugeInt
```

Construtor

Nome	Assinatura	Breve descrição
------	------------	-----------------

HugeInt	(val: Any)	Converte número ou texto para HugInt
---------	------------	--------------------------------------

ao converter de Text, a entrada é interpretada como uma representação hexadecimal

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro se zero
nonEmpty	<> Boolean	Retorna verdadeiro se não zero
size	<> Number	Retorna número de bits para representar
toText	<> Text	Retorna número em texto hexadecimal
freeze	<> Void	Torna o valor imutável
modExp	(exp: HugeInt, mod: HugeInt): HugeInt	Retorna exponenciação modular
toNumber	<> Number	Converte para número (com perda)
toChunk	<> Chunk	Codifica para representação binária

Exemplos

```
_ <- fat.type.HugeInt

# Convertendo HugeInt
x = 0x1f4 # 500 em hexadecimal
x.toText # retorna '1f4'
x.toNumber # retorna 500
x.size # 9 bits são necessários para representar 500

# Exponenciação modular
y = 0x3 # 3 em hexadecimal
z = 0x5 # 5 em hexadecimal
mod = 0x7 # 7 em hexadecimal
y.modExp(z, mod) # 0x5, equivalente a (3^5) % 7
```

Notas de uso

Conversão de Number para HugeInt

- O valor máximo permitido para conversão de Number é 2^{53} .
- Tentar passar um valor maior que 2^{53} gerará um `ValueError`.

Conversão de HugeInt para Number

- Valores até $2^{1023} - 1$ podem ser convertidos, embora alguma perda de precisão possa ocorrer para valores muito grandes.
- Se o valor exceder esse limite, o resultado será `infinity`. Isso pode ser verificado usando o método `isInf` da [biblioteca matemática](#).

a biblioteca de matemática também fornece o valor `maxInt`, que serve para avaliar a potencial perda de precisão; se um número é menor que `maxInt`, sua conversão de `HugeInt` é considerada segura sem perda de precisão

HugeInt

Veja também

- [HugeInt \(sintaxe\)](#)
- [Pacote type](#)

Text

Extensões do protótipo Text

Importação

```
_ <- fat.type.Text
```

Construtor

Nome	Assinatura	Breve descrição
Text	(val: Any)	Coage valor para texto, igual a .toText

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro se o comprimento for zero
nonEmpty	<> Boolean	Retorna verdadeiro se comprimento não-zero
size	<> Number	Comprimento do texto de retorno
toText	<> Text	Retorna o próprio valor
freeze	<> Void	Torna o valor imutável
replace	(old: Text, new: Text): Text	Substituir old por new (todos)
indexOf	(frag: Text): Number	Obter índice de fragmento, -1 se ausente
contains	(frag: Text): Boolean	Verifica se o texto contém fragmento
count	(frag: Text): Number	Obtém contagem de repetições do fragmento
startsWith	(frag: Text): Boolean	Verifica se começa com o fragmento
endsWith	(frag: Text): Boolean	Check if ends with fragment
split	(sep: Text): List/Text	Divide texto por sep em lista
toLowerCase	<> Text	Retorna a versão minúscula do texto
toUpperCase	<> Text	Retorna a versão maiúscula do texto
trim	<> Text	Retorna a versão aparada do texto
match	(re: Text): Boolean	Retorna se o texto corresponde à regex
groups	(re: Text): Scope	Retorna grupos correspondentes à regex
repeat	(n: Number): Text	Retorna o texto repetido n vezes
overlay	(base: Text, align: Text): Text	Retorna o texto sobreposto a base
patch	(i, n, val: Text): Text	Insere val na posição i, removendo n caracteres
toChunk	<> Chunk	Codifica para representação binária

Exemplo

```
_ <- fat.type.Text
x = 'banana'
x.size                # retorna 6
x.replace('nana', 'nquete'); # produz 'banquete'
```

regex

Ao definir expressões regulares, prefira usar [textos brutos](#) e lembre-se de escapar as barras invertidas conforme necessário, garantindo que as expressões regulares sejam interpretadas corretamente:

```
alphaOnly = "^[[:alpha:]]+$"
'abc'.match(alphaOnly) == true

ipAddress = "^( [0-9]{1,3} )\\. ( [0-9]{1,3} )\\. ( [0-9]{1,3} )\\. ( [0-9]{1,3} )$"
'192.168.1.2'.groups(ipAddress) == {
  _0 = '192.168.1.2'
```

Text

```
_1 = '192'  
_2 = '168'  
_3 = '1'  
_4 = '2'  
}
```

o dialeto implementado é [POSIX regex estendido](#)

overlay (sobreposição)

O valor padrão de alinhamento (se não fornecido) é 'left' (esquerda). Outros valores possíveis são 'center' (centro) e 'right' (direita):

```
'x'.overlay('___')           # 'x__'  
'x'.overlay('___', 'left')  # 'x__'  
'x'.overlay('___', 'center') # '_x_'  
'x'.overlay('___', 'right') # '___x'
```

o resultado é sempre do mesmo tamanho que o parâmetro base, o texto será cortado se for mais longo

Veja também

- [Text \(sintaxe\)](#)
- [Pacote type](#)

Method

Extensões do protótipo Method

Importação

```
_ <- fat.type.Method
```

Alias

- [Procedure](#): uma função sem argumentos que executa automaticamente quando referenciada

Construtor

Nome	Assinatura	Breve descrição
Method	(val: Any)	Envolve val em um método

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna falso, sempre
nonEmpty	<> Boolean	Retorna verdadeiro, sempre
size	<> Number	Retorna 1, sempre
toText	<> Text	Retorna o literal de texto 'Method'
freeze	<> Void	Torna o valor imutável
arity	<> Number	Retorna a aridade do método

Veja também

- [Method \(sintaxe\)](#)
- [Pacote type](#)

List

Extensões do protótipo List

Importação

```
_ <- fat.type.List
```

Construtor

Nome **Assinatura** **Breve descrição**

List (val: Any) Envolver val em uma lista

o construtor aceita um único argumento e o envolve em uma lista, se vários argumentos forem passados, apenas o primeiro argumento é considerado e os demais são ignorados; o que é consistente com o suporte a [tratamento de argumentos](#) do FatScript

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro se o comprimento for zero
nonEmpty	<> Boolean	Retorna verdadeiro para comprimento não zero
size	<> Number	Retorna o comprimento da lista
toText	<> Text	Retorna o literal de texto 'List'
freeze	<> Void	Torna o valor imutável
join	(sep: Text): Text	Junta lista com separador em texto
flatten	<> List	Achata lista de listas em uma plana
find	(p: Method): Any	Retorna a primeira correspondência ou nulo
contains	(p: Method): Boolean	Checa se algum item corresponde ao predicado
filter	(p: Method): List	Retorna sub-lista que corresponde ao predicado
reverse	<> List	Retorna uma cópia invertida da lista
shuffle	<> List	Retorna uma cópia embaralhada da lista
unique	<> List	Retorna itens únicos da lista
sort	<> List	Retorna uma cópia ordenada da lista
sortBy	(key: Any): List	Retorna uma cópia ordenada da lista *
indexOf	(item: Any): Number	Retorna índice do item, -1 se ausente
head	<> Any	Retorna o primeiro item, nulo se vazio
tail	<> List	Retorna todos os itens, exceto o primeiro
map	(m: Method): List	Utilitário funcional (permite encadeamento)
reduce	(m: Method, acc: Any): Any	Utilitário funcional
walk	(m: Method): Void	Aplicar efeitos colaterais para cada item
patch	(i, n, val: List): List	Insere val na posição i, removendo n itens
headOption	<> Option	Retorna o primeiro item, como Option
itemOption	(index: Number): Option	Retorna item por índice, como Option
findOption	(p: Method): Option	Busca item por predicado, como Option

Exemplo

```
_ <- fat.type.List
x = [ 'a', 'b', 'c' ]
x.size # retorna 3
```

Ordenação

Os métodos `sort` e `sortBy` implementam o algoritmo de ordenação quicksort, aprimorado com a seleção aleatória de pivô. Essa abordagem é conhecida por sua eficiência, oferecendo uma complexidade de tempo médio de caso de $O(n \log n)$. Demonstra alto desempenho na maioria dos conjuntos de dados. Para conjuntos de dados que contêm valores ou chaves repetidas uma ordenação estável não pode ser garantida e seu desempenho pode degradar até $O(n^2)$ no pior caso, onde todos os elementos são idênticos ou tem a mesma chave.

`sortBy` aceita um parâmetro textual para `key` se for uma lista de `Scope` ou um parâmetro numérico caso seja uma lista de `List` (matriz), representando o índice

Reduzindo

O método `reduce` no FatScript transforma uma lista em um único valor aplicando um redutor (`m: Method`) a cada elemento em sequência, começando de um valor acumulador inicial (`acc: Any`), ou do primeiro elemento caso nenhum valor seja provido. Este método é útil para operações que envolvem agregar dados de uma lista.

Características

- **Método Redutor:** O redutor deve receber o valor acumulador atual e o item atual da lista, retornando o valor acumulador atualizado.
- **Comportamento com Lista Vazia:** Quando o `reduce` é aplicado a uma lista vazia sem um valor acumulador inicial, ele retorna `null`.

restrição: o método redutor será considerado inválido se não receber dois parâmetros ou caso defina valores padrão para eles

Exemplo prático

```
_ <- fat.type.List
sumReducer = (acc: Número, item: Número) -> acc + item
sum = [1, 2, 3].reduce(sumReducer) # resulta em 6
```

para transformações complexas de dados ou ao lidar com listas de escopos, estruture cuidadosamente o redutor para manipular os tipos de dados específicos e o resultado desejado

Veja também

- [List \(sintaxe\)](#)
- [Tipo Option](#)
- [Pacote type](#)

Scope

Extensões do protótipo Scope

Importação

```
_ <- fat.type.Scope
```

Alias

Keyset: uma lista de chaves (List/Text)

Construtor

Nome	Assinatura	Breve descrição
Scope	(val: Any)	Envolver val em um escopo

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro se o tamanho for zero
nonEmpty	<> Boolean	Retorna verdadeiro para tamanho não zero
size	<> Number	Retorna o número de entradas
toText	<> Text	Retorna o literal de texto 'Scope'
freeze	<> Void	Torna o valor imutável
seal	<> Void	Impede o crescimento do escopo
isSealed	<> Boolean	Verifica se o escopo está selado
copy	<> Scope	Retorna cópia profunda do escopo
keys	<> Keyset	Retorna lista de chaves do escopo
valuesOf	(t: Type): List	Obtém valores correspondentes ao tipo t
pick	(keys: Keyset): Scope	Filtra o escopo pelas chaves
omit	(keys: Keyset): Scope	Filtra o escopo removendo as chaves
vmap	(m: Method): Scope	Mapeia valores (m = v1 -> v2)
maybe	(key: Text): Option	Retorna valor dentro de Option

Exemplo

```
_ <- fat.type.Scope
x = { num = 12, prop = 'outra' }
x.size # retorna 2
```

Veja também

- [Scope \(sintaxe\)](#)
- [Tipo Option](#)
- [Pacote type](#)

Error

Extensões do protótipo Error

Importação

```
_ <- fat.type.Error
```

Aliases

- AssignError: atribuindo um novo valor a uma entrada imutável
- AsyncError: falha na operação assíncrona
- CallError: uma chamada é feita com argumentos insuficientes
- FileError: falha na operação de arquivo
- IndexError: o índice está fora dos limites da lista/texto
- KeyError: a chave (nome) não é encontrada no escopo
- SyntaxError: erro de sintaxe ou estrutura de código
- TypeError: inconsistência de tipo em chamada, retorno ou atribuição de método
- ValueError: tipo pode estar correto, mas conteúdo não é aceito

Construtor

Nome	Assinatura	Breve descrição
Error	(val: Any)	Retornar val coagido para texto como erro

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro, sempre
nonEmpty	<> Boolean	Retorna falso, sempre
size	<> Number	Retorna 0, sempre
toText	<> Text	Retorna o texto do erro
freeze	<> Void	Torna o valor imutável

Exemplo

```
_ <- fat.type.Error

# Gerando um erro intencionalmente
x = Error('ops')
x.toText # retorna "Error: ops"

# Causando um erro inadvertidamente
e = undeclared.item # causa um TypeError
e.toText           # retorna "TypeError: can't resolve scope of 'item'"
```

Aliases de erro na prática

```
# Exemplo de AssignError
x = 10
x = 20 # gera "AssignError: reassignment to immutable > x"

# Exemplo de IndexError
list = [ 1, 2, 3 ]
list[5] # gera "IndexError: out of bounds"

# Exemplo de CallError
add(10) # gera "CallError: nothing to call > add > ..."
```

Veja também

Error

- [Biblioteca failure](#)
- [Error \(sintaxe\)](#)
- [Pacote type](#)

Chunk

Extensões do protótipo Chunk

Importação

```
_ <- fat.type.Chunk
```

Alias

- `ByteArray`: uma sequência de bytes (Number [0-255])

Construtor

Nome	Assinatura	Breve descrição
Chunk	(val: Any)	Coage valor para bloco (binário)

se o valor for do tipo `Number`, cria um bloco de n bytes inicializados com zero, onde n é o número fornecido

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>isEmpty</code>	<> Boolean	Retorna verdadeiro se o tamanho for zero
<code>nonEmpty</code>	<> Boolean	Retorna verdadeiro se tamanho não-zero
<code>size</code>	<> Number	Retorna o tamanho do bloco (em bytes)
<code>toText</code>	<> Text	Converte o bloco para formato de texto
<code>freeze</code>	<> Void	Torna o valor imutável
<code>toBytes</code>	<> ByteArray	Converte o bloco para uma lista de bytes
<code>toHugeInt</code>	<> HugeInt	Cria HugeInt a partir de dados binários
<code>seek</code>	(frag: Chunk, offset: Number = 0): Number	Retorna índice da primeira correspondência
<code>seekByte</code>	(byte: Number, offset: Number = 0): Number	Retorna índice da primeira correspondência
<code>patch</code>	(i, n, val: Chunk): Chunk	Insere val na posição i, removendo n bytes
<code>fit</code>	(len: Number): Chunk	Trunca para um comprimento fixo

`toText` substitui quaisquer sequências inválidas de UTF-8 por U+FFFD, representado como  em UTF-8

Exemplo

```
_ <- fat.type.Chunk

# Criando um bloco a partir de texto
x = Chunk('example')

x.size      # 7, o tamanho em bytes
x.toText    # 'example', representado como texto
x.toBytes   # [ 101, 120, 97, 109, 112, 108, 101 ], os valores UTF-8

x.seek(Chunk('am'))      # 2, a posição da correspondência
x.patch(1, 5, Chunk('XY')) # um novo bloco 'eXYe'

# Criando um bloco a partir de um número
y = Chunk(5) # cria um bloco de 5 bytes inicializados com zero

y.size      # retorna 5
y.toBytes   # retorna [ 0, 0, 0, 0, 0 ]
```

Veja também

- [Chunk \(sintaxe\)](#)

Chunk

- [Pacote type](#)

extra._

Tipos adicionais implementados em FatScript puro:

- [Date](#) - Gerenciamento de calendário e datas
- [Duration](#) - Construtor de duração em milissegundos
- [Fuzzy](#) - Valores probabilísticos e operações de lógica fuzzy
- [HashMap](#) - Armazenamento rápido de chave-valor
- [Logger](#) - Suporte ao registro de logs
- [Memo](#) - Utilitário de memoização genérica
- [MouseEvent](#) - Parser de eventos de mouse
- [Opaque](#) - Utilitário para proteção suave de dados
- [Option](#) - Encapsulamento de valor opcional
- [Param](#) - Verificação de presença e tipo de parâmetro
- [Sound](#) - Interface de reprodução de som
- [Storable](#) - Armazenamento de dados

Importando

Se você quiser disponibilizar todos eles de uma só vez, basta escrever:

```
_ <- fat.extra._
```

...ou importe um por um, conforme necessário, por exemplo:

```
_ <- fat.extra.Date
```

Date

Gerenciamento de calendário e datas

operações como adição e subtração de dias, meses e anos, garantindo o tratamento preciso de várias complexidades relacionadas a datas, como anos bissextos e cálculos de final de mês

Importação

```
_ <- fat.extra.Date
```

[biblioteca time](#), [biblioteca math](#), [tipo Error](#), [tipo Text](#), [tipo List](#), [tipo Number](#), [tipo Duration](#) são automaticamente importados com esta importação

Tipo Date

Date oferece uma solução abrangente para o gerenciamento de datas, incluindo anos bissextos e horário do dia.

Propriedades

- `year`: Número - Ano da data
- `month`: Número - Mês da data
- `day`: Número - Dia da data
- `tms`: Milissegundos - Horário do dia em milissegundos

valor padrão aponta para: 1 de janeiro de 1970

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>fromEpoch</code>	(ems: Epoch): Date	Cria uma instância a partir de um epoch
<code>isLeapYear</code>	(year: Número): Boolean	Determina se um ano é bissexto
<code>normalizeMonth</code>	(month: Número): Número	Normaliza o número do mês
<code>daysInMonth</code>	(year: Número, month: Número): Número	Retorna o número de dias no mês de um ano
<code>isValid</code>	(year, month, day, tms): Boolean	Valida os componentes da data
<code>truncate</code>	<> Date	Trunca o horário do dia
<code>toEpoch</code>	<> Epoch	Converte a instância para tempo em epoch
<code>addYears</code>	(yearsToAdd: Número): Date	Adiciona anos à data
<code>addMonths</code>	(monthsToAdd: Número): Date	Adiciona meses à data
<code>addWeeks</code>	(weeksToAdd: Número): Date	Adiciona semanas à data
<code>addDays</code>	(daysToAdd: Número): Date	Adiciona dias à data

Exemplos de uso

```
_ <- fat.extra.Date

# Criar uma instância de Data
myDate = Date(2023, 1, 1)

# Adicionar um ano à data
newDate = myDate.addYears(1)

# Adicionar duas semanas à uma data
datePlusTwoWeeks = myDate.addWeeks(2)

# Criar uma Data a partir de um tempo em epoch (em milissegundos)
# o resultado é influenciado pelo fuso horário atual, veja: time.setZone
epochTime = 1672531200000
dateFromEpoch = Date.fromEpoch(Epoch(epochTime))
```

Date

```
# Converter uma data para tempo em epoch  
epochFromDate = myDate.toEpoch
```

Duration

Construtor de duração em milissegundos

no FatScript, o tempo é nativamente expresso em milissegundos, e esse tipo fornece uma maneira simples de expressar diferentes magnitudes de tempo em `Millis`

Importação

```
_ <- fat.extra.Duration
```

Construtor

Nome	Assinatura	Breve descrição
<code>Duration</code>	<code>(val: Number)</code>	Cria um conversor de duração em <code>Millis</code>

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>nanos</code>	<code><> Millis</code>	Interpretar valor como nanossegundos
<code>micros</code>	<code><> Millis</code>	Interpretar valor como microssegundos
<code>millis</code>	<code><> Millis</code>	Interpretar valor como milissegundos
<code>seconds</code>	<code><> Millis</code>	Interpretar valor como segundos
<code>minutes</code>	<code><> Millis</code>	Interpretar valor como minutos
<code>days</code>	<code><> Millis</code>	Interpretar valor como dias
<code>weeks</code>	<code><> Millis</code>	Interpretar valor como semanas

Exemplo

```
_ <- fat.extra.Duration
time <- fat.time

cincoSegundos = Duration(5).segundos
time.wait(cincoSegundos) # pausa a execução do thread por 5 segundos
```


Fuzzy

Valores probabilísticos e operações de lógica fuzzy

Importar

```
_ <- fat.extra.Fuzzy
```

Construtor

Nome	Assinatura	Breve descrição
Fuzzy	(val: Number = 0.5)	Cria um valor de probabilidade Fuzzy

a faixa de 0 a 1 é ideal para valores, porém, valores mais altos ainda podem ser significativos em operações específicas como conjunção com valores dentro da faixa padrão

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Verifica se a probabilidade é zero
notEmpty	<> Boolean	Verifica se a probabilidade é maior que zero
size	<> Number	Converte o valor fuzzy para uma escala percentual
toText	<> Text	Converte o valor fuzzy para uma porcentagem textual
freeze	<> Void	Torna o valor imutável
and	(other: Fuzzy): Fuzzy	Operação lógica AND com outro valor fuzzy
or	(other: Fuzzy): Fuzzy	Operação lógica OR com outro valor fuzzy
not	<> Fuzzy	Operação lógica NOT, invertendo a chance
decide	<> Boolean	Decide um resultado booleano dentro de sua chance

Uso

```
_ <- fat.extra.Fuzzy

# Criando instâncias fuzzy
lowChance = Fuzzy(0.25) # chance de 25%
highChance = Fuzzy(0.75) # chance de 75%

# Aplicando operações lógicas
combinedChance = lowChance.and(highChance)
resolvedChance = combinedChance.decide # resulta em um booleano
```

Inspiração

A introdução do tipo Fuzzy no FatScript foi inspirada pela definição humorística da linguagem meme, [DreamBerd](#), que oferece booleanos que podem ser *verdadeiro*, *falso* ou *talvez*. Aqui, a palavra-chave *talvez* é traduzida para `Fuzzy()`. `decide`, o que pode ser considerado uma construção incomum para a maioria das linguagens de programação e é análogo a jogar uma moeda.

Embora o FatScript não seja tão esotérico a ponto de armazenar booleanos como "um bit e meio", o conceito de fornecer um tipo "engraçado" que permite modelar incertezas foi um experimento interessante e pode realmente ser útil em muitos cenários. Ele melhora a capacidade da linguagem de lidar com operações envolvendo chances e processos de tomada de decisão onde os resultados não são determinísticos. O tipo Fuzzy é útil para cenários que requerem uma abordagem com nuances da lógica booleana, comumente vista na lógica de jogos e em qualquer lugar onde decisões probabilísticas sejam necessárias.

Veja também

- [Tipo Booleano](#)
- [Biblioteca de matemática](#)

HashMap

Armazenamento otimizado em memória de par chave-valor, servindo como um substituto de melhor desempenho para a implementação padrão do Scope, projetado para lidar eficientemente com grandes conjuntos de dados.

os ganhos de velocidade vem em detrimento de um maior uso de memória

Importação

```
_ <- fat.extra.HashMap
```

Construtor

Nome	Assinatura	Breve descrição
HashMap	(capacity: Number = 97)	Cria um HashMap com uma capacidade especificada

a capacidade padrão de 97 é geralmente eficiente para até 10.000 itens

Otimização de Capacidade

Idealmente, você deve manter no máximo cerca de 100 itens por 'compartimento' na tabela hash. Neste contexto, 'capacidade' refere-se ao número de compartimentos disponíveis para seus dados. Note que esta implementação não ajusta seu tamanho automaticamente, portanto, um dimensionamento inicial adequado é crucial. A seguinte tabla pode ajudar a determinar a capacidade ótima para armazenar n itens:

```
n < 5000 => 53
n < 10000 => 97
n < 20000 => 193
n < 40000 => 389
n < 80000 => 769
n < 160000 => 1543
_ => 3079
```

usar números primos pode ajudar a reduzir colisões

Estes valores são baseados em testes empíricos e devem ser ajustados de acordo com suas necessidades específicas de dados e objetivos de desempenho. Tenha em mente que a relação entre capacidade e desempenho não é totalmente linear; à medida que o número de itens aumenta, os benefícios de aumentar ainda mais a capacidade diminuem.

Recomendação

Embora o Scope padrão do FatScript mostre um desempenho mais lento para inserções, ele se destaca na recuperação e atualização de dados, superando a velocidade do HashMap para pequenas coleções (menos de ~500 itens). Portanto, os benefícios de usar o HashMap são mais notáveis em cenários que envolvem inserções frequentes em grandes conjuntos de dados.

Membros do protótipo

Nome	Assinatura	Breve descrição
isEmpty	<> Boolean	Retorna verdadeiro se o comprimento for zero
nonEmpty	<> Boolean	Retorna verdadeiro para comprimento não zero
size	<> Number	Retorna o comprimento da tabela hash
toText	<> Text	Retorna o literal de texto 'HashMap/capacity'
freeze	<> Void	Torna o valor imutável
set	(key: Text, value: Any): Any	Define um par chave-valor no HashMap
get	(key: Text): Any	Obtém o valor associado a uma chave
keys	<> Keyset	Retorna uma lista de todas as chaves do HashMap

Exemplo

HashMap

```
_ <- fat.extra.HashMap  
  
hmap = HashMap()  
hmap.set('key1', 'value1')  
  
hmap.get('key1') # retorna 'value1'  
hmap.keys       # retorna [ 'key1' ]
```

Logger

Suporte ao registro de logs

desde logs simples em console a registro baseado em arquivos

Importação

```
_ <- fat.extra.Logger
```

[biblioteca console](#), [biblioteca color](#), [biblioteca file](#), [biblioteca time](#), [biblioteca sdk](#) e [biblioteca de tipos](#) são automaticamente importadas com esta importação

Tipo Logger

Logger oferece capacidades de registro de logs personalizáveis com vários níveis e formatos.

Propriedades

- `level`: Text (padrão 'debug') - Nível de log
- `showTime`: Boolean (padrão verdadeiro) - Indicador para exibir carimbos de hora

níveis válidos: 'debug', 'info', 'warn', 'error'

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>setLevel</code>	(level: Text)	Define o nível de log
<code>setShowTime</code>	(showTime: Boolean)	Alterna a exibição de carimbos de hora nos logs
<code>asMessage</code>	(level: Text, args: Scope): Texto	Formata mensagem de log (pode ser substituído)
<code>log</code>	(msg: Any, fg: Number)	Registra mensagem (pode ser substituído)

Métodos de log

- `debug(_1, _2, _3, _4, _5)`: Registra uma mensagem de debug
- `info(_1, _2, _3, _4, _5)`: Registra uma mensagem informativa
- `warn(_1, _2, _3, _4, _5)`: Registra uma mensagem de aviso
- `error(_1, _2, _3, _4, _5)`: Registra uma mensagem de erro

Subtipos

FileLogger

- Herda de Logger
- Propriedades Adicionais:
 - `logfile`: Texto (padrão 'log.txt') - arquivo para registro de logs
- Substitui `log` para anexar mensagens a um arquivo

Exemplo de uso

```
_ <- fat.extra.Logger

# Crie uma instância com configurações personalizadas
myLogger = Logger(level = 'info', showTime = false)

# Registra uma mensagem informativa
myLogger.info('Esta é uma mensagem informativa.')

# Crie um FileLogger para registrar mensagens em um arquivo
fileLogger = FileLogger('meuLog.txt')
fileLogger.info('Registrado no arquivo.')
```

Memo

Utilitário de memoização genérica (também pode criar valores preguiçosos)

Importação

```
_ <- fat.extra.Memo
```

Construtor

Nome	Assinatura	Breve descrição
Memo	(method: Method)	Cria uma instância Memo para um método
		a aridade do método memoizado pode ser 1 ou então 0 (para valor preguiçoso)

Membros do protótipo

Nome	Assinatura	Breve descrição
asMethod	<> Method	Retorna uma versão memoizada do método original
call	(arg: Any): Any	Chamada memoizada; armazena e retorna resultado

Exemplo

Memo é útil para otimizar funções, armazenando os resultados. Ela armazena o resultado das chamadas de função e retorna o resultado armazenado quando as mesmas entradas ocorrem novamente.

```
_ <- fat.extra.Memo

fib = (n: Number) -> {
  n <= 2 => 1
  _      => quickFib(n - 1) + quickFib(n - 2)
}

quickFib = Memo(fib).asMethod

quickFib(50) # 12586269025
```

Agora você pode chamar `quickFib` como se estivesse chamando `fib`, mas com resultados armazenados em cache para entradas computadas anteriormente.

aviso: pode causar acúmulo na alocação de memória

MouseEvent

Parser de eventos de mouse para `fat.curses.readKey`

Importação

```
_ <- fat.extra.MouseEvent
```

Construtor

Nome	Assinatura	Breve descrição
MouseEvent	(val: Text)	Faz o parse de um evento readKey

O construtor `MouseEvent` recebe o seguinte argumento:

- **val**: O valor de texto retornado por `fat.curses.readKey`, que é interpretado para extrair dados do evento de mouse (como tipo de clique, posição e estado de liberação).

se `val` não for um evento de mouse válido, `MouseEvent` retornará `null`

Membros do protótipo

Nome	Assinatura	Breve descrição
name	<> Text	Retorna nome legível do evento de mouse
code	Text	O código que representa a ação do mouse
x	Number	A coordenada X do mouse
y	Number	A coordenada Y do mouse
isRelease	Boolean	O evento é uma liberação de botão?

Nomeação dos eventos

O método `name` pode retornar os seguintes valores legíveis, com base no código do evento e nas teclas modificadoras:

- **Ações de Pressionar**: `leftPress`, `middlePress`, `rightPress`
- **Ações de Soltar**: `leftRelease`, `middleRelease`, `rightRelease`
- **Ações de Arrastar**: `leftDrag`, `middleDrag`, `rightDrag`
- **Ações de Rolagem**: `scrollUp`, `scrollDown`
- **Modificadores**: `Shift+`, `Alt+`, `Ctrl+` (prefixados às ações acima)

Exemplo

`MouseEvent` é útil para converter eventos brutos em dados utilizáveis, como a posição do mouse e a ação realizada:

```
_ <- fat.extra.MouseEvent
console <- fat.console
curses <- fat.curses

# Ativar o rastreamento do mouse
curses.setMouse(true)

# Capturar eventos do curses
(~ key = curses.readKey) == Void @ {
  # Aguarde até que um evento ocorra... (no-op)
}

# Fazer o parse do evento do curses
(mEvt = MouseEvent(key)) => {
  console.log('Evento de mouse:')
  console.log('  Código: {mEvt.code}')
  console.log('  X: {mEvt.x}')
  console.log('  Y: {mEvt.y}')
```

MouseEvent

```
    console.log(' Liberado: {mEvt.isRelease}')
    console.log(' Ação: {mEvt.name}')
  }
  _ => console.log('Evento não relacionado ao mouse: {key}')
```

Veja também

- [Biblioteca curses](#)

Opaque

Utilitário para proteção suave de dados

Importação

```
_ <- fat.extra.Opaque
```

o [tipo Scope](#) é automaticamente importado junto com esta importação.

Protótipo

Esta biblioteca introduz o tipo `Opaque`, projetado como um utilitário para encapsular dados com proteção suave. Ele fornece um invólucro em torno de um `SCOPE`, impedindo o acesso direto à estrutura subjacente quando serializando, mas permitindo interações com seus membros.

O tipo `Opaque` estende o tipo `Scope`, herdando todos os seus membros sem introduzir recursos adicionais.

Exemplo de uso

```
visible = { ~ a = 1, b = 2 }  
  
# Criando um invólucro opaco  
opaque = Opaque(visible)  
  
# Modificando valores no escopo oculto  
opaque.c = 3 # funciona como um Scope comum  
  
# O escopo oculto é excluído da serialização no nível superior  
recode.toJSON({ opaque, visible }) # '{"opaque":null,"visible":{"a":1,"b":2}}'  
  
# A serialização direta do escopo oculto ainda funciona  
recode.toJSON(opaque) # '{"a":1,"b":2,"c":3}'
```

Notas de desempenho

O tipo `Opaque` introduz uma camada de indireção que pode impactar o desempenho em cenários que envolvem acesso ou modificações frequentes nos dados ocultos. Considere usá-lo de forma seletiva para casos que exigem encapsulamento, por exemplo, ao armazenar credenciais em uma instância.

Veja também

- [Tipo Scope](#)
- [Biblioteca Recode](#)

Option

Encapsulamento de valor opcional

Importação

```
_ <- fat.extra.Option
```

o [tipo Error](#) é automaticamente importado junto com esta importação

Tipos

Esta biblioteca introduz dois principais tipos: `Some` e `None`, que são casos especiais do tipo `Option`, fornecendo uma maneira de representar valores opcionais, encapsulando a presença (`Some`) ou ausência (`None`) de um valor.

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>isEmpty</code>	<code><> Boolean</code>	Verifica se a opção é <code>None</code>
<code>nonEmpty</code>	<code><> Boolean</code>	Verifica se a opção é <code>Some</code>
<code>size</code>	<code><> Number</code>	Retorna 1 se <code>Some</code> , 0 se <code>None</code>
<code>toText</code>	<code><> Text</code>	Retorna o literal de texto
<code>freeze</code>	<code><> Void</code>	Torna o valor imutável
<code>get</code>	<code><> Any</code>	Retorna valor ou erro <code>NoSuchElement</code>
<code>getOrElse</code>	<code>(default: Any): Any</code>	Retorna valor ou default se for <code>None</code>
<code>map</code>	<code>(fn: Method): Option</code>	Aplica o método ao valor contido
<code>flatMap</code>	<code>(fn: Method/Option): Option</code>	Aplica método que retorna <code>Option</code>
<code>filter</code>	<code>(predicate: Method): Option</code>	Filtra o valor pelo predicado
<code>toList</code>	<code><> List</code>	Converte a opção para Lista
<code>concrete</code>	<code><> Option</code>	Resolve a opção para <code>Some</code> ou <code>None</code>

Exemplo de uso

```
_ <- fat.extra.Option

# Criando opções
x = Some(5) # equivalente a Option(5).concrete
y = None() # equivalente a Option().concrete

# Trabalhando com opções
isEmptyX = x.isEmpty # false
isEmptyY = y.isEmpty # true
valX = x.getOrElse(0) # 5
valY = y.getOrElse(0) # 0

# Aplicando uma transformação
transformedX = x.map(v -> v * 2).getOrElse(0) # 10
transformedY = y.map(v -> v * 2).getOrElse(0) # 0

# Elevando valores a opção
label: Text = Option(opVal).concrete >> {
  Some => 'algum valor' # caso onde opVal não é null
  None => 'nenhum valor' # caso onde opVal é null
}
```

Option em Programação Funcional

No `FatScript`, `null` é integrado como um cidadão de primeira classe, permitindo que, na maioria dos casos, tipos nativos manipulem valores ausentes sem a necessidade de construtos adicionais para segurança. Consequentemente, o tipo `Option` está incluído no pacote `extra` como açúcar sintático.

Option

Ele permite a encapsulação explícita de valores opcionais para clareza semântica ou aderência a certos paradigmas de programação funcional. Um exemplo de sua utilidade é demonstrado no tipo `Scope`, que inclui um método `maybe` além da sintaxe padrão de recuperação de valor:

- `myScope('key')` retorna o valor associado a `key` ou `null` se a chave não existir.
- `myScope.maybe('key')` fornece um valor envolto em `Option`, distinguindo explicitamente entre a existência (`Some`) e a ausência (`None`) de um valor.

Manipulação semântica de valores ausentes

Um dos principais benefícios de usar o tipo `Option` é sua capacidade de lidar semanticamente e com segurança com operações que envolvem valores potencialmente ausentes. Esse recurso é particularmente útil em operações primitivas ou transformações de dados onde valores `null` poderiam, de outra forma, levar a erros. Por exemplo, considere um cenário onde você precisa somar um número com um valor que pode não estar presente:

```
# Supondo que ovosComprados esteja definido e tenha um valor
ovosComprados: Number = ...
```

```
# geladeira.maybe('ovo') recupera o número de ovos na geladeira como uma Option
# Se 'egg' não estiver presente, o padrão é 0, evitando erros relacionados a null
totalDeOvos: Number = geladeira.maybe('ovo').getOrElse(0) + ovosComprados
```

Considerações de desempenho

O uso de tipos `Option` introduz sobrecarga computacional devido às chamadas de função necessárias para manipular valores e à memória adicional decorrente de sua estrutura subjacente. Embora os benefícios de segurança e expressividade sejam significativos, o custo de desempenho pode se tornar perceptível em loops apertados ou ao processar grandes conjuntos de dados.

Veja também

- [Tipo Scope](#)
- [Tipo Error](#)

Param

Verificação de presença e tipo de parâmetro

Importação

```
_ <- fat.extra.Param
```

o [tipo Erro](#) é automaticamente importado com esta importação

Tipos

Esta biblioteca introduz o tipo `Param` e a utilidade `Using/UsingStrict` para declaração de parâmetros implícitos.

Construtores

Tanto `Param` quanto `Using/UsingStrict` recebem dois argumentos:

- **_exp: Text:** o nome do parâmetro a ser verificado no contexto.
- **_typ: Type:** o tipo esperado do valor avaliado.

Adicionalmente, `Param` aceita um argumento opcional:

- **strict: Boolean:** desativa a [correspondência flexível](#) (padrão é falso).

`Using` tem essa flag como `false` e `UsingStrict` como `true`

Param

O tipo `Param` oferece mecanismos para verificar a presença e o tipo de parâmetros no contexto de execução.

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>get</code>	<code><> Any</code>	Recupera o parâmetro se corresponder ao tipo

o método `get` gera `KeyError` se o parâmetro não estiver definido, e `TypeError` se o tipo não corresponder

Exemplo

```
_ <- fat.extra.Param
_ <- fat.type.Text # o tipo desejado deve ser carregado

currentUser = Param('userId', Text)
```

...

```
# Supondo que userId esteja definido no contexto e seja um texto,
# recupere seu valor de forma segura do namespace atual
userId = currentUser.get
```

Using

Aplique `Using/UsingStrict` para suprimir dicas de parâmetros implícitos em declarações de métodos para entradas esperadas no escopo.

alternativamente, para suprimir avisos sobre parâmetros implícitos, nomeie a entrada implícita começando com um sublinhado (`_`)

Exemplo

```
_ <- fat.extra.Param
_ <- fat.type.Text
```

Param

```
printUserIdFromContext = <> {  
    Using('userId', Text)  
    console.log(userId)  
}
```

se o parâmetro implícito estiver ausente ou não corresponder, um erro será gerado em tempo de execução quando o método for chamado

Veja também

- [Pacote extra](#)

Sound

Interface de reprodução de som

wrapper para players de áudio de linha de comando usando [fork e kill](#)

Importação

```
_ <- fat.extra.Sound
```

Construtor

O construtor `Sound` recebe três argumentos:

- **path**: o caminho do arquivo de áudio.
- **duração** (opcional): o tempo de espera (em milissegundos) para poder reproduzir novamente o arquivo. Geralmente, você deseja definir isso como a duração exata do seu áudio.
- **player** (opcional): o player padrão utilizado é `aplay` (utilitário de áudio comum no Linux, apenas dá suporte a wav), mas você pode usar `ffplay` para reproduzir mp3, por exemplo, definindo `ffplay = ['ffplay', '-nodisp', '-autoexit', '-loglevel', 'quiet']` e fornecendo-o como argumento para sua instância de som. Neste caso o pacote `ffmpeg` precisa estar instalado no sistema.

Membros do protótipo

Nome	Assinatura	Breve descrição
<code>play</code>	<code><> Void</code>	Inicia reprodução, se já não estiver tocando
<code>stop</code>	<code><> Void</code>	Interrompe reprodução, se ainda estiver tocando

o estado de "estar tocando" é inferido pelo parâmetro de duração

Exemplo

```
_ <- fat.extra.Sound
time <- fat.time

applause = Sound('applause.wav', 5000);
applause.play
time.wait(5000)
```

note que `Sound` cria um processo filho para reproduzir o áudio, portanto, a reprodução é assíncrona

Som na Web Build

Ao usar `fry` compilado com Emscripten (por exemplo, ao usar [FatScript Playground](#)), este protótipo utiliza comandos embutidos `$soundPlay` e `$soundStop`, que são definidos apenas na compilação para web. Portanto, em vez de utilizar um reprodutor de áudio CLI através de `fork` do processo, há suporte de áudio via `SDL2/WebAudio`.

Veja também

- [Pacote extra](#)

Storable

Armazenamento de dados

Importação

```
_ <- fat.extra.Storable
```

[biblioteca file](#), [biblioteca sdk](#), [biblioteca enigma](#), [tipo Error](#), [tipo Text](#), [tipo Void](#) e [tipo Method](#) são automaticamente importados com esta importação

Mixins

Esta biblioteca introduz dois tipos de mixin: `Storable` e `EncryptedStorable`

Storable

O mixin `Storable` fornece métodos para armazenar e recuperar objetos no sistema de arquivos usando serialização JSON.

Membros do protótipo

Nome	Assinatura	Breve descrição
list	<> Keyset	Obtém lista de ids para instâncias armazenadas
load	(id: Text): Any	Carrega um objeto do sistema de arquivos
save	<> Boolean	Salva a instância do objeto atual
erase	<> Boolean	Exclui o arquivo associado ao id

os métodos `load` e `save` emitem `FileError` em caso de falha

EncryptedStorable

Estende `Storable` com capacidades de criptografia para um armazenamento de dados mais seguro. Requer uma implementação do método `getEncryptionKey`.

Exemplo de uso

```
_ <- fat.extra.Storable

# Defina um tipo que inclua Storable (ou EncryptedStorable)
User = (
  Storable # Inclui o mixin Storable

  # EncryptedStorable # implementação alternativa
  # getEncryptionKey = (): Text -> '3ncryp1ptM3' # poderia obter via KMS ou
  configuração

  ## Argumentos
  name: Text
  email: Text

  # Os setters retornam uma nova cópia imutável da instância com o campo atualizado
  setName = (name: Text) -> self + User * { name }
  setEmail = (email: Text) -> self + User * { email }
)

# Cria uma nova instância de usuário
newUser = User('Jane Doe', 'jane.doe@example.com')

# Salva o novo usuário
newUser.save

# Atualiza as informações do usuário e salva as alterações
```

Storable

```
updatedUser = newUser
  .setName('Jane Smith')
  .setEmail('jane.smith@example.com')
updatedUser.save

# Lista todos os usuários salvos
userIds = User.list

# Carrega um usuário do sistema de arquivos
userId = userIds(0) # ...ou newUser.id
loadedUser = User.load(userId)

# Exclui os dados do usuário do sistema de arquivos
loadedUser.erase # ...ou User.erase(userId)
```

Storable na Web Build

Ao usar fry compilado com Emscripten (por exemplo, ao usar [FatScript Playground](#)), este protótipo utiliza comandos embutidos `$storableSave`, `$storableLoad`, `$storableList` e `$storableErase`, que são definidos apenas na compilação para web. Portanto, em vez de utilizar o sistema de arquivos convencional para o armazenamento, há suporte especial para utilização do objeto `localStorage` do navegador.

Veja também

- [Pacote extra](#)

Comandos embutidos

Comandos embutidos são funções de baixo nível do FatScript que podem ser invocadas com palavras-chave precedidas por um cifrão \$. Esses comandos estão sempre disponíveis, implementados como código compilado e não requerem importações.

Ao contrário dos métodos, eles não recebem argumentos explícitos, mas podem ler a partir de nomes de entrada específicos no escopo atual ou até mesmo do estado interno do interpretador.

Os mais úteis

Aqui estão alguns comandos embutidos que podem valer a pena conhecer:

- `$break` pausa a execução e carrega o console de depuração
- `$debug` alterna os logs de depuração do interpretador (apenas em algumas builds)
- `$exit` encerra o programa com o código fornecido
- `$keepDotFry` mantém a config (.fryrc) no escopo após a inicialização
- `$result` alterna a impressão do resultado no final da execução
- `$root` fornece uma referência ao escopo global
- `$self` fornece uma referência própria ao escopo do método/instância
- `$bytesUsage` retorna o máximo de bytes alocados
- `$nodesUsage` retorna o total de nós alocados no momento
- `$isMain` verifica se o código está sendo executado como principal ou módulo
- `$port` recupera número da porta efetivamente atribuída ao servidor HTTP

as palavras-chave `root` e `self` são automaticamente convertidas em `$root` e `$self`

Você pode chamá-los diretamente no seu código, assim:

```
$exit # encerra o programa
```

para usar outros comandos embutidos você deve estudar a implementação C de fry, já que a lista completa não está documentada, consulte o arquivo [embedded.c](#)

Bibliotecas por trás dos bastidores

As bibliotecas padrão embalam chamadas embutidas como métodos, fornecendo uma interface mais ergonômica. Você não precisa criar um escopo de execução ou carregar argumentos nesse escopo antes de delegar a execução a eles.

Por exemplo, veja como você pode usar o método `floor` da [biblioteca math](#):

```
_ <- fat.math
floor(2.53)
```

Este método é implementado como:

```
floor = (x: Number): Number -> $floor
```

Por trás dos bastidores, o método `floor` cria um escopo de execução e carrega um argumento como `x` nele. O método então delega a execução ao comando embutido `$floor`, que por sua vez, lê o valor de `x` do escopo atual e retorna o menor inteiro que não é maior que este número.

Você pode obter o mesmo resultado que o método acima fazendo o seguinte:

```
x = 2.53
$floor # lê o valor de x do escopo atual
```

Hackeando

Você pode ver qual comando embutido um método de biblioteca está chamando, olhando para a implementação da biblioteca através do método `readLib` da [biblioteca SDK](#). Tecnicamente, não há nada que o impeça de chamar comandos embutidos diretamente.

Por exemplo, você pode encerrar seu programa chamando `$exit` diretamente, que sairá com o código 0 (padrão) ou, se uma entrada numérica chamada `code` existir no escopo atual, o valor dessa entrada será usado como código de saída. No entanto, seria mais elegante importar a biblioteca `fat.system` e chamar o método `exit` com o código de saída desejado:

Comandos embutidos

```
sys <- fat.system  
sys.exit(0) # exits with code 0
```

Essa abordagem torna seu código mais legível e menos propenso a erros, além de fornecer uma melhor separação de responsabilidades.

É importante ter em mente que os comandos embutidos são caixas pretas e não são destinados à escrita de código FatScript comum. Na maioria dos casos, você precisaria ler a [implementação em C subjacente](#) para entender melhor o que um comando está realmente fazendo.

Embora seja possível usar comandos embutidos para obter desempenho adicional em tempo de execução, evitando importações e chamadas de método, isso não é recomendado devido à perda de legibilidade do código. Em geral, é melhor usar as bibliotecas padrão e seguir as melhores práticas para escrever um código claro, fácil de manter.